SECOND FEATURE

SUBROUTINE MASTER

DOS 3.3 O 0

Now you can build a library of Applesoft subroutines that behave much like Pascal procedures. Features include two-way parameter passing, local variables, nesting, and recursion.

ProDOS O 0

by H. Cem Kaner and John R. Vokey, 256C Calle Marguerita, Los Gatos, CA 95030

One of Applesoft's main drawbacks is the lack of "real" subroutines. With BASIC's weak subroutine capabilities, we find ourselves constantly rewriting the same sections of code, and tailoring them to each new program. Ideally, we should be writing fairly general subroutines, recording them on a subroutine library disk, and then merging them with any program that needs them. Subroutine libraries save time and reduce errors: once a general-purpose subroutine has been debugged, it can be used with confidence, and without error, in program after program.

Despite the advantages of library creation, we find that we don't do this in standard Applesoft, nor do we often swap subroutines or take subroutines from articles without modifying them, often extensively. In contrast, even in a language as antiquated and downright clumsy as FORTRAN II, developing subroutine libraries is a very natural programming activity. We also have problems translating programs written in the usual business and scientific languages into Applesoft. Most of the published programs worth translating rely heavily on subroutine facilities that do not exist in BASIC. It's not impossible to get around this, but it makes translation a frustrating, time-consuming, and error-ridden process.

Subroutine Master (**Listing 1**) adds "real" subroutine handling to Applesoft. It's still not perfect and program execution is slower than we would like. But using this program has saved us a great deal of programming and debugging time, which easily justifies the cost of some of the computer's (not our) execution time.

AN EXAMPLE

The program shown in **Example 1** is a very simple example of subroutine calls using our handler. This program uses the same subroutine, SRT, to sort the elements of two different arrays into ascending order. A normal subroutine (**line 150**) is used to print the unsorted X() array in **line 40** and the sorted values in **line 60**. We'll discuss the features and syntax of Subroutine Master's subroutines in detail shortly. For now, read through the remarks, which illustrate some of the handler's capabilities. Note the following features of Subroutine Master:

1. *Reference to a subroutine by name.* The subroutine named SRT starts at **line 90**. It wouldn't matter if this were moved. The computer will find the subroutine SRT wherever it appears in the program.
2. *Variable passing to the subroutine.* **Line 60** CALLs (GOSUBs) SRT and passes two pieces of information to it. As called from **line 60**, SRT sorts the ten elements of array X(). When called from **line 70**, SRT sorts the 20 elements of array Y() instead. In both cases, SRT thinks it's working with N elements of array S(). It has no idea that these are called by different names in the main (calling) program.
3. *Local variables.* In **line 100**, variables I, J and S are declared LOCAL to SRT. This means that they are created specifically for the SRT routine, and that they will only exist in memory while SRT is active. The value of the main program's variable, I, which was used in **lines 40 and 50**, is absolutely unaffected by any changes in the value of the local variable I. The two I's have the same name, but they are entirely different variables.

USING THE PROGRAM

Before you use any of the commands described below, SUBR.MASTER (**Listing 1**) must be installed and the beginning of the Applesoft program adjusted upward. See the two demonstration programs (**Listings 2 and 3**) for examples of how to do this from within an Applesoft program. The demonstration programs will run as they are, as long as a disk with SUBR.MASTER is in the current disk drive.

Near the beginning of the program, the variable EXIT should be set to 4058, and each subroutine name should be set to 3141. (Note that Applesoft only distinguishes variables by their first two characters. However, Subroutine Master can distinguish longer names. The two-character variable should not be changed from the initial 3141 setting.)

Five statements are included in the Subroutine Master system:

CALL ***name,parameter list*** — This calls a subroutine by *name*, passing the variables or expressions in the *parameter list* to the

EXAMPLE 1: A Program Using Subroutine Master

```
                                                    Remarks
10  IF  PEEK (104) <  > 17 THEN  POKE 103,44:       Reload program and load subroutine master
    POKE 104,17: POKE 4395,0: PRINT  CHR$ (
    4)"BLOAD SUBR.MASTER": PRINT  CHR$ (4)"R
    UN EXAMPLE1"
20  SRT = 3141:EXIT = 4058                          Handler address definitions
30  DIM X(10),Y(20)
40  FOR I = 1 TO 10:X(I) =  RND (1): NEXT : GOSUB 150   Starts with data in random order
50  FOR I = 1 TO 20:Y(I) =  RND (1): NEXT
60  CALL SRT,X(0),10: GOSUB 150                     Sorts it
70  CALL SRT,Y(0),20                                Uses CALL statement and variable list
80  END
90  DEF SRT,S(0),N                                  Subroutine DEFinition
100 LOCAL,I,J,S                                     Declaration of LOCAV variables
110 FOR I = 1 TO N - 1: FOR J = I + 1 TO N          Actual subroutine starts here
120 IF S(I) > S(J) THEN S = S(I):S(I) = S(J)        Reorders the array elements from smallest to
    :S(J) = S                                       largest
130 NEXT J,I
140 CALL EXIT,SRT                                   EXIT (subroutine return) statement
150 FOR I = 1 TO 10: PRINT X(I):NEXT:RETURN         Standard subroutine to print array X()
```

corresponding variables in the subroutine's DEF header. Floating point, integer, and string variables, as well as arrays of all three types, may be included in the *parameter list*. The CALL *name* statement may appear anywhere in an Applesoft program line. **Line 140** of **Listing 2** is an example of passing a string literal, while **line 350** demonstrates passing a floating point variable.

DEF *name,parameter list* — This marks the beginning of a named subroutine. The variables in the *parameter list* receive values from items in the corresponding *parameter list* of a CALL statement. The variables used in the *parameter list* are local variables, which are passed back to the corresponding CALL variables on return from the subroutine. The DEF statement must be the first statement on a program line. **Line 390** of **Listing 1** is a typical DEF statement.

CALL EXIT,*name* — This marks the end of a named subroutine. The *name* used in the DEF header must be included in the CALL EXIT statement. CALL EXIT must be the last statement on a program line. **Line 420** of **Listing 2** is the CALL EXIT statement that corresponds to **line 390**.

CALL DISP,*variable list* — DISP (short for DISPOSE) removes the variables named in the *variable list* from memory. If you need to use this command, you must also set DISP equal to 2304 at the beginning of the program.

LOCAL,*variable list* — This optional statement must be the next statement after the DEF statement of the subroutine. It declares the variables in the *variable list* as local variables — distinct from variables of the same name used in the main program or in other subroutines. **Line 780** of **Listing 2** makes the variable Z$ local to the RET subroutine.

Values are passed from the parameters listed in the CALL *name* statement to the variables listed in the DEF statement. If the CALL *name* statement uses variables, rather than expressions, the values of DEF variables are passed back to variables in the CALL statement.

An entire array may be passed as a parameter in a CALL *name* statement. This is indicated by simply placing a zero in place of each index. The corresponding array variable in the DEF statement should be indicated in the same way. You can pass an array element to a subroutine as a simple variable, but you can't send an array element or a simple variable to an element of a subroutine array. Arrays may be specified as LOCAL by specifying the array in the same way you would in a DIM statement. The LOCAL statement automatically dimensions the array.

The system does not allow you to create new global variables inside a subroutine. If you want to change a global variable from within a subroutine, make sure that it has already been created in the main program.

User-defined functions (DEF FN) may not be included in a CALL *name* statement, nor may they be used inside a subroutine. Other restrictions, idiosyncracies, and error messages are discussed later.

ENTERING THE PROGRAMS

To key in SUBR.MASTER, either use your assembler to enter the source code from **Listing** 1, or type CALL −151 <RETURN> and use the Monitor to enter the hex codes. The entire source file may be too long to fit into the memory available with some assemblers. In this case, you may have to split it into two parts, as we did with Apple's DOS Tool Kit Assembler. Be sure that the name of the second file and the name specified in the CHN (or equivalent) command at the end of the first file match. Then save the program with the command:

BSAVE SUBR.MASTER,A$900,L$82B

Key in **Listing 2** and save it with the command:

SAVE SUBR.MAST.DEMO1

Key in **Listing 3** and save it with the command:

SAVE SUBR.MAST.DEMO2

These programs relocate themselves in memory, so it is important that you save them before you run them. Also, be sure that SUBR.MASTER is on the disk in the current disk drive. For help in entering *Nibble* listings, see "A Welcome to New *Nibble* Readers" at the beginning of this issue.

DESIGN CRITERIA

What features should "real" subroutines have? We knew how we wanted the subroutine handler to interact with the user long before we figured out how to achieve this. We worked on a number of conceptually very different approaches to creating "real" subroutines before settling on the one presented here. In this section we describe our general goals and outline the approaches taken to meet them.

Non-Interference With GOSUB

Nothing in our program affects GOSUB, POP or RETURN in any way, and we do not store our return parameters in the stack, which is GOSUB and FOR territory. GOSUB subroutines can still be used freely and will come in handy in many programs.

One Entry, One Exit

You should always have to enter a subroutine at the same place, the beginning, and leave it at the same place, the end. This is a key restriction underlying the philosophies of modular and structured programming, mainly because it eliminates a regular source of programming errors.

Our program does not allow multiple entry points in a subroutine. However, multiple exit statements are possible within subroutines. Since multiple entry points and unexpected subroutine entries caused us much more grief than multiple exits, we were less worried about restricting the exits.

Space Efficiency

In passing variables from the main program to a subroutine, as we did with the

SRT routine, we rename X() to S(), calling it X() again on exit. This is fast and simple, unless there's already an array called S() in memory. In that case, the new S() (the old X()) must be moved down in memory so that the new S() will be used in SRT. Instead of costing us 5,000 or more bytes (as it would if we had chosen to copy the array), passing arrays costs us no memory beyond the length of the routines required to rename, check, and, if necessary, move the variables.

A second space-expensive trick is to set aside a reserved area of memory for local variables. This approach allows local variables to retain their values, but it adds the requirement of zeroing these variables and it consumes way too much memory. This is a luxury — if it is a luxury — that we simply cannot afford. Instead, we get rid of the locals on exit from subroutines, freeing up the memory they occupied for use by the rest of the program.

In both cases, we trade speed for space efficiency. We are much more concerned about handling lots of data, running large programs, and using high resolution graphics than we are (usually) about saving a few seconds. When speed is more important, we can always use the old standby, custom-tailored GOSUB subroutines, instead.

Recursion

In recursive programming languages (such as Pascal), subroutines can call themselves freely. In contrast, languages like FORTRAN never allow a subroutine to call itself. Applesoft is partially recursive. Up to the limits of available space in the stack (which disappears fairly quickly), subroutines can call themselves. We felt that our routines should be as fully recursive as possible, treating all of free memory as a stack.

In principle our subroutines allow extensive recursion. In practice, with CALL piled upon CALL, highly recursive programs run quite slowly, which limits the utility of this approach. Still, it will be important for some users, particularly students who wish to learn about recursion, that such programs execute correctly, if not promptly. **Listing 3** is a simple example of using recursion.

Variable Passing

In **Example 1**, we called SRT twice, once passing it array X(), the next time passing it Y(). To do this in standard BASIC, in **line 60** we would have had to resort to something like:

FOR I=1 TO 10: S(I)=X(I): NEXT: N=10: GOSUB 100

instead of:

CALL SRT,X(),10

Then at **line 70** we would have had to do the same thing again just to pass down Y() and 20. This is a tedious and error-prone method.

LISTING 1: SUBR.MASTER

```
0000:            1 ;**********************
0000:            2 ;*    SUBR.MASTER     *
0000:            3 ;*  BY CEM KANER AND  *
0000:            4 ;*    JOHN VOKEY      *
0000:            5 ;* COPYRIGHT (C) 1985 *
0000:            6 ;* BY MICROSPARC, INC *
0000:            7 ;* CONCORD, MA  01742 *
0000:            8 ;**********************
0000:            9 ;
0000:           10 ; DOS TOOLKIT ASSEMBLER
0000:           11 ;
0000:           12 ;      PROGRAM CONSTANTS
0000:           13 ;
0024:           14 STRING  EQU  $24        ; $
0025:           15 PERCENT EQU  $25        ; %
0028:           16 LPAREN  EQU  $28        ; (
0029:           17 RPAREN  EQU  $29        ; )
002C:           18 COMMA   EQU  $2C        ; ,
003A:           19 COLON   EQU  $3A        ; :
0041:           20 EH      EQU  $41        ; A
0043:           21 CE      EQU  $43        ; C
004C:           22 EL      EQU  $4C        ; L
004F:           23 OH      EQU  $4F        ; O
008C:           24 CALL    EQU  $8C        ; CALL TOKEN
00B8:           25 DEF     EQU  $B8        ; DEF TOKEN
0000:           26 ;
0000:           27 ;      APPLESOFT ROUTINES
0000:           28 ;
00B1:           29 CHRGET  EQU  $B1        ; FETCH CHR AT TXTPTR
00B7:           30 CHRGOT  EQU  $B7        ; RECOVER LAST CHR, SET FLAGS
D393:           31 BLTU    EQU  $D393      ; BLOCK TRANSFER UP
D39A:           32 BLTUP   EQU  $D39A      ; BLTU AFTER REASON TEST
D412:           33 ERROR   EQU  $D412      ; DIE HORRIBLY
D410:           34 OMERR   EQU  $D410      ; OUT OF MEMORY
D697:           35 STXTPT  EQU  $D697      ; TXTPTR = START OF PROGRAM TEXT
D998:           36 ADDON   EQU  $D998      ; ADD Y TO TXTPTR
D9A6:           37 REMN    EQU  $D9A6      ; PUT OFFSET TO EOL IN REG Y
DA52:           38 LETCNT  EQU  $DA52      ; LATE ENTRY TO LET
DB97:           39 GETTXT  EQU  $DB97      ; TRANSFER OLDTXT TO TXTPTR
DD76:           40 MISMATCH EQU $DD76      ; TYPE MISMATCH
DEBE:           41 CHKCOM  EQU  $DEBE      ; CRASH IF NOT COMMA
DEC9:           42 SYNERR  EQU  $DEC9      ; SYNTAX ERROR
DFD9:           43 DIM     EQU  $DFD9      ; DIMENSION COMMAND
DFE3:           44 PTRGET  EQU  $DFE3      ; FIND VARIABLE IN MEMORY
E07D:           45 ISLETC  EQU  $E07D      ; SET CARRY IF A HOLDS A LETTER
E199:           46 QUANTERR EQU $E199      ; ILLEGAL QUANTITY ERROR
E1BC:           47 DATAERR EQU  $E1BC      ; OUT OF DATA
E306:           48 ERRDIR  EQU  $E306      ; CRASH IF IN IMMEDIATE MODE
0000:           49 ;
0000:           50 ;      APPLESOFT ADDRESSES
0000:           51 ;
0010:           52 DIMFLAG EQU  $10        ; DIMENSION FLAG
0011:           53 VALTYP  EQU  $11        ; FF IF STRING
0012:           54 INTFLAG EQU  $12        ; 80 IF INTEGER
0014:           55 SUBFLAG EQU  $14        ; 80 IF SUBSCRIPT OK
003C:           56 MOVESTART EQU $3C       ; MONITOR A1L, A1H
003E:           57 MOVEND  EQU  $3E        ; MONITOR A2L, A2H
0042:           58 MOVETO  EQU  $42        ; MONITOR A4L, A4H
0069:           59 VARTAB  EQU  $69        ; START VARIABLE STORAGE
006B:           60 ARYTAB  EQU  $6B        ; START ARRAY STORAGE
006D:           61 STREND  EQU  $6D        ; END VARIABLE STORAGE
0075:           62 CURLIN  EQU  $75        ; CURRENT LINE #
0079:           63 OLDTXT  EQU  $79        ; OLD TEXT POINTER
0081:           64 LASTVAR EQU  $81        ; LATEST VARIABLE NAME
0083:           65 VARPNT  EQU  $83        ; POINTS LATEST VARIABLE VALUE
0085:           66 FORPNT  EQU  $85        ; USED BY LET
0094:           67 HIGHDS  EQU  $94        ; HIGH DESTINATION, BLTU
0096:           68 HIGHTR  EQU  $96        ; HIGH TRANSFER, BLTU
009B:           69 LOWTR   EQU  $9B        ; PTR TO VAR NAME OR LOW TRANSFER
00B8:           70 TXTPTR  EQU  $B8        ; TEXT POINTER
0200:           71 BUFR    EQU  $200       ; INPUT BUFFER
0000:           72 ;
0801:           73 PO      EQU  $801       ; TRUE PROGRAM ORIGIN
----- NEXT OBJECT FILE NAME IS XX
0900:           74         ORG  PO+$FF     ; 1ST PAGE FOR DATA
0900:           75 ;
0900:           76 ;      PROGRAM VARIABLES
0900:           77 ;
0900:           78 ; $77-DF LOCS USED ARE WALKED ON BY APPLESOFT
0900:           79 ; ERROR, BUT DON'T INTERFERE WITH APPLESOFT
0900:           80 ; FUNCTIONING ITSELF, SO ARE SAFE TEMPS.
0900:           81 ;
0077:           82 NUMCHR  EQU  $77        ; (OLDLIN) # CHRS TO MOVE
0078:           83 NUMPAGE EQU  $78        ; # PAGES TO MOVE
00DA:           84 BUFPTR  EQU  $DA        ; (ERRLIN) PTR FOR BUFR, SECBUF
00DB:           85 COUNTER EQU  $DB        ; (ERRLIN,ERRPOS) 2ND PTR
00DD:           86 ARYFLAG EQU  $DD        ; (ERRPOS) FF IF ARRAY OR ARY EXPR
00DE:           87 EXPRFLAG EQU $DE        ; (ERRNUM) FF IF EXPR
00DF:           88 PARENCOUNT EQU $DF      ; (ERRSTK) # PARENS LEFT
00FA:           89 GENPTR  EQU  $FA        ; GENERAL POINTER
00FC:           90 CALLPTR EQU  $FC        ; POINT TO CALL LIST
00FE:           91 DEFPTR  EQU  $FE        ; POINT TO DEF LIST
0801:           92 FATOFF  EQU  PO         ; STORE OLD $FA TO $FF
0807:           93 DEFLIST EQU  PO+$06     ; POINT TO START DEF VARLIST
0809:           94 CALLIST EQU  PO+$08     ; POINT TO START CALL VARLIST
```

```
080B:                95 DEFLINE EQU  PO+$0A     ; DEF LINE NUMBER
080D:                96 PROCNAME EQU PO+$0C     ; POINT TO PROC NAME
080F:                97 OLDARYTAB EQU PO+$0E    ; SAVED ARYTAB
0811:                98 OLDVARTAB EQU PO+$10
0813:                99 OLDSTREND EQU PO+$12
0815:               100 OLDSIMPLE EQU PO+$14    ; START OF ORIGINAL SIMPLES
0817:               101 NEWARYTAB EQU PO+$16    ; FOR MOVE ROUTINE
0819:               102 HOLDCOMMA EQU PO+$18    ; SAVES CHR FROM PUTCOLON
081A:               103 SECBUF  EQU  PO+$19     ; SECOND BUFFER
00D0:               104 BUFMAX  EQU  $D0        ; MAX # CHRS ALLOWED IN SECBUF
0900:               105         CHN  SUBR.MAST.S2
0900:                 1 ;
0900:                 2 ;     SUBROUTINES
0900:                 3 ;
0900:                 4 ***********************************
0900:                 5 *    CALL THIS ADDRESS TO DISPOSE   *
0900:                 6 *    OF A VARIABLE.  FOR DOCUMEN-   *
0900:                 7 *    TATION SEE VOKEY & KANER, 1982 *
0900:                 8 ***********************************
0900:                 9 ; NOTE: THIS ROUTINE IS MODIFIED FROM THE
0900:                10 ; BYTE PAPER AS FOLLOWS: JSR MOVE ($FE2C)
0900:                11 ; IS CHANGED TO JSR NEWMOVE ($096F).  THIS
0900:                12 ; CUTS EXECUTION TIME FOR LARGE MOVES BY UP
0900:                13 ; TO 70% BUT ELIMINATES RELOCATABILITY
0900:                14 ; OF THE CODE.
0900:                15 ;
0900:20 B1 00        16 DISPOSE JSR  CHRGET     ; MOVE PAST COMMA
0903:20 B7 00        17 CLEAR   DFB  $20,$B7,$00,$D0,$03,$4C,$6C,$D6
0906:D0 03 4C
0909:6C D6
090B:20 E3 DF        18         DFB  $20,$E3,$DF,$C4,$6C,$D0,$02,$C5
090E:C4 6C D0
0911:02 C5
0913:6B A0 02        19         DFB  $6B,$A0,$02,$08,$B0,$0A,$A9,$00
0916:08 B0 0A
0919:A9 00
091B:C8 91 9B        20         DFB  $C8,$91,$9B,$88,$A9,$07,$91,$9B
091E:88 A9 07
0921:91 9B
0923:18 B1 9B        21         DFB  $18,$B1,$9B,$85,$44,$A5,$9B,$85
0926:85 44 A5
0929:9B 85
092B:42 65 44        22         DFB  $42,$65,$44,$85,$3C,$A5,$9C,$85
092E:85 3C A5
0931:9C 85
0933:43 C8 71        23         DFB  $43,$C8,$71,$9B,$85,$3D,$B1,$9B
0936:9B 85 3D
0939:B1 9B
093B:85 45 A0        24         DFB  $85,$45,$A0,$00,$A5,$6D,$85,$3E
093E:00 A5 6D
0941:85 3E
0943:A5 6E 85        25         DFB  $A5,$6E,$85,$3F,$20,$6F,$09,$A5
0946:3F 20 6F
0949:09 A5
094B:6D E5 44        26         DFB  $6D,$E5,$44,$85,$6D,$A5,$6E,$E5
094E:85 6D A5
0951:6E E5
0953:45 85 6E        27         DFB  $45,$85,$6E,$28,$B0,$0A,$A5,$6B
0956:28 B0 0A
0959:A5 6B
095B:E9 06 85        28         DFB  $E9,$06,$85,$6B,$B0,$02,$C6,$6C
095E:6B B0 02
0961:C6 6C
0963:20 B7 00        29         DFB  $20,$B7,$00,$D0,$01,$60,$20,$BE
0966:D0 01 60
0969:20 BE
096B:DE 38 B0        30         DFB  $DE,$38,$B0,$9C
096E:9C
096F:38              31 NEWMOVE SEC             ; SET UP IN SAME WAY AS
0970:A5 3E           32         LDA  MOVEND     ; MONITOR MOVE BUT EXECUTION
0972:E5 3C           33         SBC  MOVESTART  ; IS MUCH FASTER FOR
0974:85 77           34         STA  NUMCHR     ; NON-TRIVAL MOVES.  MOVEND
0976:A5 3F           35         LDA  MOVEND+1   ; MUST BE STRICTLY GREATER
0978:E5 3D           36         SBC  MOVESTART+1 ; THAN MOVESTART.
097A:85 78           37         STA  NUMPAGE    ; # FULL PAGES TO MOVE
097C:B0 03           38         BCS  ADD1       ; DO A RANGE CHECK.  CRASH
097E:4C 99 E1        39         JMP  QUANTERR   ; IF MOVEND <= MOVESTART.
0981:E6 77           40 ADD1    INC  NUMCHR     ; TOTAL # BYTES IS 1 SHY
0983:D0 02           41         BNE  PAGECHECK  ; FROM SUBTRACTION.  SO ADD
0985:E6 78           42         INC  NUMPAGE    ; IT BACK IN
0987:A5 78           43 PAGECHECK LDA NUMPAGE   ; ANY FULL PAGES TO MOVE?
0989:F0 11           44         BEQ  PARTMOVE   ; IF NOT.  DO PARTIAL PAGE
098B:A0 00           45         LDY  #0         ; START OF FULL PAGE
098D:B1 3C           46 PAGEMOVE LDA (MOVESTART),Y ; MOVES
098F:91 42           47         STA  (MOVETO),Y
0991:C8              48         INY             ; PAGE DONE?
0992:D0 F9           49         BNE  PAGEMOVE   ; LEAVES WITH Y=0
0994:E6 3D           50         INC  MOVESTART+1 ; ADJUST FOR NEXT PAGE
0996:E6 43           51         INC  MOVETO+1
0998:C6 78           52         DEC  NUMPAGE    ; ANOTHER LEFT?
099A:D0 F1           53         BNE  PAGEMOVE   ; DO TILL DONE LAST.
099C:A5 77           54 PARTMOVE LDA NUMCHR     ; ANY LEFT?
099E:F0 09           55         BEQ  MOVEDONE   ; CARRY SET FROM BCS ADD1
09A0:B1 3C           56 PARTMOVE1 LDA (MOVESTART),Y ; MOVE LAST NUMCHR BYTES
09A2:91 42           57         STA  (MOVETO),Y ; Y STARTS 0 FROM ABOVE
09A4:C8              58         INY             ; RUNS TO NUMCHR-1
09A5:C4 77           59         CPY  NUMCHR     ; FOR TOTAL OF NUMCHR BYTES
```

Other languages solve the problem of passing variables by forcing the programmer to specify which variables a subroutine is to act on, each time the routine is called. So we allow variable lists as in FORTRAN, Pascal, COBOL, and many other languages. There is one big difference, of course. If you want to use a GOSUB with no explicit variable passing, you can use a GOSUB. But now you don't have to.

Ideally, variable passing should be as unrestricted as is variable assignment in Applesoft. You should be able to pass reals to integers and vice versa; to pass array elements (like X(10)) back and forth; and to pass expressions to simple variables. We achieved most, but not all of this.

Portability

If you can use a friend's subroutine correctly in your own program after spending less than five minutes examining it, the subroutine is "portable" — it moves easily from person to person and from program to program. Making it easy to write portable subroutines is the main goal of this program. A variety of factors increase portability. We've looked at one already: the less a subroutine is tied to specific variable names, the more general, and the more portable, it will be.

Named Subroutines — People understand names much better than they understand line numbers when they are trying to figure out the function of some section of a program. Accordingly, it should be possible to refer to the separate sections of a program by name, rather than location.

Our program searches for actual names, without ever using line numbers, by scanning the start of each line for a DEF token. When it finds a DEF, it compares what follows to the name of the subroutine it's looking for. When it finds a matching name, it has found the right routine.

Reusable Variable Names — When was the last time you wrote something like:

FOR I = 1 TO 10: GOSUB 1000

only to discover later that the subroutine at **line 1000** changes the value of I? This kind of bug is as annoying as it is common. You should be able to write a subroutine without worrying about what variable names will be used in any of the programs that call it. The subroutine's variables should not affect those of the main program unless you want them to.

Our first step in eliminating conflicts between main and subprogram variable names was to create local variables. Declare a variable LOCAL in a subroutine and a brand new variable (any type, simple or array) of this name is created in memory. Reference to this variable has no effect on any of the main program's data. Further, because the locals are cleared out of memory on exit from the subroutine, that routine

gives back as much free memory as it got, so there's no conflict with future main program variable storage requirements either.

As a second step, we added variable passing. The variables passed to a subroutine are renamed to the names in the DEF list. We make sure that variables passed to the subroutine are stored lower in memory than variables of the main program, which have the same names as those found in the DEF lists. Because of this, Applesoft always operates, in the subroutine, on the subroutine's variables. This protects the main program's variables from being changed in the subroutine accidentally. Thus, you can send variables to the subroutine without knowing or caring what it will call them there; you can call them whatever you want in the subroutine. You will affect the variables you think you are working with, and no others.

A third level of protection against unexpected reference to variables in a subroutine was built in to allow natural use of "global" variables. A global variable is defined in the main program but can be used in a subroutine without appearing in the DEF or LOCAL parameter lists. Some (not many) variables can and should be safely made global. Think of D$=CHR$(4), for example. DOS requires the programmer to define this, or something like it, in every program that uses disk access. It's tedious enough doing DOS's housekeeping for it once per program, so you shouldn't have to worry about passing it or redefining it for each subroutine. Globals should be made and unmade in the main program; subroutines that tinker with global storage are not portable. This system detects the initial definition of a global variable within a subroutine, and signals it with a MEMORY ERROR.

Explicit Subroutine Interface — To use a subroutine correctly, you need to understand its inputs (what gets passed down), its outputs (what variables it can and does affect) and its function. These three aspects (the "subroutine interface" with the main program) are all that you need to know about the subroutine. You do not need to know the details (the "subroutine quagmire") of how the subroutine does what it's supposed to do. As long as it does it correctly, don't worry about how it does it.

If the subroutine interface is laid out clearly, correctly and briefly, you should be able to use that routine in your program within minutes. If the interface has to be fished out of the quagmire, you may as well, and probably will, rewrite the beast instead.

For a subroutine to be portable, it must be well documented, meaning that its interface must be easy to find and understand. The DEF statement's variable list tells you automatically what types of variables the subroutine expects as input. If no globals are used, the DEF statement describes the types of all inputs. The CALL statement's variable list identifies the inputs themselves, tell-

LISTING 1: SUBR.MASTER *(continued)*

```
09A7:90 F7      60          BCC  PARTMOVE1
09A9:60         61 MOVEDONE RTS             ; LEAVES WITH CARRY SET
09AA:           62 ;
09AA:           63 ;     HOUSEKEEPING
09AA:           64 ;
09AA:20 06 E3   65 IN       JSR  ERRDIR     ; USES 200+. CRASH IN IMM MODE
09AD:A2 05      66          LDX  #5         ; TRANSFER FA TO FF
09AF:B5 FA      67 SAVEPØ   LDA  $FA,X      ; TO A SAFE PLACE
09B1:9D Ø1 Ø8   68          STA  FATOFF,X   ; WILL PUT THEM BACK
09B4:CA         69          DEX             ; ON EXIT
09B5:10 F8      70          BPL  SAVEPØ     ; LOOP TILL DONE
09B7:60         71          RTS
09B8:A2 05      72 OUT      LDX  #5         ; RESTORE FA TO FF
09BA:BD Ø1 Ø8   73 BACKPØ   LDA  FATOFF,X   ; TO THEIR
09BD:95 FA      74          STA  $FA,X      ; OLD HOME
09BF:CA         75          DEX
09C0:10 F8      76          BPL  BACKPØ
09C2:60         77          RTS
09C3:A5 B8      78 SAVETXT  LDA  TXTPTR     ; SAVE TEXT POINTER
09C5:85 79      79          STA  OLDTXT     ; IN APPLESOFT'S
09C7:A5 B9      80          LDA  TXTPTR+1   ; USUAL
09C9:85 7A      81          STA  OLDTXT+1   ; HIDEYHOLE
09CB:60         82          RTS
09CC:A5 FE      83 POINTDEF LDA  DEFPTR     ; POINT TEXT POINTER
09CE:85 B8      84          STA  TXTPTR     ; AT THE DEF LIST
09DØ:A5 FF      85          LDA  DEFPTR+1
09D2:85 B9      86          STA  TXTPTR+1
09D4:60         87          RTS
09D5:A5 FC      88 POINTCALL LDA CALLPTR    ; POINT TXTPTR
09D7:85 B8      89          STA  TXTPTR     ; AT THE CALL LIST
09D9:A5 FD      90          LDA  CALLPTR+1
09DB:85 B9      91          STA  TXTPTR+1
09DD:60         92          RTS
09DE:A5 B8      93 TXTTODEF LDA  TXTPTR     ; UPDATE
09EØ:85 FE      94          STA  DEFPTR     ; DEF POINTER
09E2:A5 B9      95          LDA  TXTPTR+1
09E4:85 FF      96          STA  DEFPTR+1
09E6:60         97          RTS
09E7:A5 B8      98 TXTTOCALL LDA TXTPTR     ; UPDATE
09E9:85 FC      99          STA  CALLPTR    ; THE CALL PTR
09EB:A5 B9     100          LDA  TXTPTR+1
09ED:85 FD     101          STA  CALLPTR+1
09EF:60        102          RTS
09FØ:AD Ø9 Ø8  103 STARTLIST LDA CALLIST    ; POINT THE
09F3:85 FC     104          STA  CALLPTR    ; CALL AND DEF PTRS
09F5:AD ØA Ø8  105          LDA  CALLIST+1  ; AT THE START OF
09F8:85 FD     106          STA  CALLPTR+1  ; THEIR VARIABLE
09FA:AD Ø7 Ø8  107          LDA  DEFLIST    ; LISTS
09FD:85 FE     108          STA  DEFPTR
09FF:AD Ø8 Ø8  109          LDA  DEFLIST+1
0A02:85 FF     110          STA  DEFPTR+1
0A04:60        111          RTS
0A05:AD ØD Ø8  112 POINTNAME LDA PROCNAME   ; POINT
0A08:85 B8     113          STA  TXTPTR     ; TO PROC NAME
0A0A:AD ØE Ø8  114          LDA  PROCNAME+1
0A0D:85 B9     115          STA  TXTPTR+1
0A0F:60        116          RTS
0A10:20 Ø5 ØA  117 GETNAME  JSR  POINTNAME  ; POINT TO PROC NAME
0A13:20 B1 ØØ  118          JSR  CHRGET     ; ADVANCE PAST LEADING COMMA
0A16:4C E3 DF  119          JMP  PTRGET     ; FIND IT IN MEMORY
0A19:A5 B8     120 DECTXT   LDA  TXTPTR     ; MOVE TXTPTR
0A1B:DØ Ø2     121          BNE  DECTXTLOW  ; BACK 1
0A1D:C6 B9     122          DEC  TXTPTR+1   ; FROM HIGH BYTE IF NEEDED
0A1F:C6 B8     123 DECTXTLOW DEC TXTPTR    ; FROM LOW BYTE ALWAYS
0A21:60        124          RTS
0A22:A5 69     125 SAVEVARTAB LDA VARTAB    ; THESE ROUTINES
0A24:8D 11 Ø8  126          STA  OLDVARTAB  ; USED TO STASH
0A27:A5 6A     127          LDA  VARTAB+1   ; APPLESOFT POINTERS
0A29:8D 12 Ø8  128          STA  OLDVARTAB+1 ; WHILE THE ROUTINE
0A2C:60        129          RTS             ; PLAYS WITH THEM
0A2D:AD 11 Ø8  130 GETVARTAB LDA OLDVARTAB  ; ALL ARE MADE
0A30:85 69     131          STA  VARTAB     ; SUBROUTINES
0A32:AD 12 Ø8  132          LDA  OLDVARTAB+1 ; NO MATTER HOW
0A35:85 6A     133          STA  VARTAB+1   ; OFTEN CALLED
0A37:60        134          RTS
0A38:A5 6B     135 SAVEARYTAB LDA ARYTAB    ; BECAUSE THEY
0A3A:8D ØF Ø8  136          STA  OLDARYTAB  ; ARE OF GENERAL USE
0A3D:A5 6C     137          LDA  ARYTAB+1   ; FOR MANY DIFFERENT
0A3F:8D 1Ø Ø8  138          STA  OLDARYTAB+1 ; UTILITY PROGRAMS
0A42:60        139          RTS
0A43:AD ØF Ø8  140 GETARYTAB LDA OLDARYTAB  ; SINCE THIS
0A46:85 6B     141          STA  ARYTAB     ; PROGRAM WILL ALMOST
0A48:AD 1Ø Ø8  142          LDA  OLDARYTAB+1 ; ALWAYS BE IN CORE
0A4B:85 6C     143          STA  ARYTAB+1   ; (FOR US ANYWAY)
0A4D:60        144          RTS             ; THIS GIVES US A
0A4E:A5 6D     145 SAVESTREND LDA STREND    ; STANDARD PLACE
0A50:8D 13 Ø8  146          STA  OLDSTREND  ; TO FIND THEM
0A53:A5 6E     147          LDA  STREND+1   ; INSTEAD OF
0A55:8D 14 Ø8  148          STA  OLDSTREND+1 ; REWRITING THEM
0A58:60        149          RTS             ; DOZENS OF TIMES
0A59:AD 13 Ø8  150 GETSTREND LDA OLDSTREND
0A5C:85 6D     151          STA  STREND
0A5E:AD 14 Ø8  152          LDA  OLDSTREND+1
0A61:85 6E     153          STA  STREND+1
0A63:60        154          RTS
0A64:20 22 ØA  155 SIMPTOVAR JSR SAVEVARTAB ; PUT OLDSIMPLE
```

ing you automatically, for each call of the subroutine, which variables are inputs and which variables (the whole list) are possible outputs.

If no globals are used, this list covers all possible outputs of the subroutine. No other variables can be changed by it. Add a few REMs on function and variable use at the top of the subroutine, and your documentation of the interface is complete.

Even better, the most important part of this documentation, the statement of inputs and outputs, is always correct. Because the variable lists of the DEF and CALL statements are part of the code, they can never misrepresent the code, as REMs sometimes do. There is no better (or easier) form of documentation than self-documentation, automatically generated by writing the code.

THE FINER POINTS

Passing Simple Variables

When a simple variable is encountered in the DEF list, a new local variable is automatically created with this name and is set equal to the variable or expression in the CALL list. Anything that Applesoft considers legal to do for an equal sign (=) is legal for a pass. You can't pass strings to integers, or commands to anything (X= GOTO?), but otherwise passing is quite flexible. Each simple variable in the DEF list adds seven bytes of overhead to variable storage while the subroutine is in effect. DEF simple variables are erased from memory on exit from the subroutine, after their values have been passed back to the CALL variables. When a large amount of data is stored in memory, creating and clearing these local variables can take noticeable amounts of time.

User-Defined Functions

Functions such as FN A(X) should never be passed to or used within a subroutine. This is the sole exception to the '=' rule for simple variables. Applesoft's internal handling of functions makes use of the absolute memory location of the function, but this location is changed when local variables are created and can be changed when variables are erased. Subroutine calls always involve local variable creation, for storing return pointers (we have to put them somewhere, and as we don't use the stack, we stash them in local variables having the same name as the subroutine). Functions are almost always mishandled as a result. Subroutines exit with simple variable storage as they found it, so functions defined in the main program can be used in the main program at any time.

Passing Arrays

Array variables are always passed via renaming. The array named in the CALL list is given the DEF list name on the way to the subroutine, and it is given its old name back on exit. Typically, this requires only a few microseconds, no matter how large

LISTING 1: SUBR.MASTER *(continued)*

```
0A67:AD 15 08   156          LDA  OLDSIMPLE ; IN THE SIMPLE PTR
0A6A:85 69      157          STA  VARTAB    ; WITHOUT LOSING
0A6C:AD 16 08   158          LDA  OLDSIMPLE+1 ; VARTAB'S OLD
0A6F:85 6A      159          STA  VARTAB+1  ; VALUE
0A71:60         160          RTS
0A72:AD 0B 08   161 DEFTOCUR LDA  DEFLINE   ; PUT THE DEF STATEMENT
0A75:85 75      162          STA  CURLIN    ; LINE NUMBER IN CURLIN
0A77:AD 0C 08   163          LDA  DEFLINE+1
0A7A:85 76      164          STA  CURLIN+1
0A7C:60         165          RTS
0A7D:           166 ;
0A7D:           167 ;     VARIABLE HANDLING SUBROUTINES
0A7D:           168 ;
0A7D:A0 00      169 SKIPVAR LDY  #0         ; BYPASS THIS VAR OR EXPRN
0A7F:84 DF      170          STY  PARENCOUNT ; COUNTS PARENTHESES
0A81:C8         171 SKIP1    INY            ; ON ENTRY TXTPTR POINTS
0A82:B1 B8      172          LDA  (TXTPTR),Y ; AT LEADING COMMA
0A84:F0 21      173          BEQ  SKIPPED   ; LEAVE ON END OF LINE
0A86:C9 3A      174          CMP  #COLON    ; SAME AS END OF LINE
0A88:F0 1D      175          BEQ  SKIPPED
0A8A:C9 2C      176          CMP  #COMMA    ; END OF VARIABLE?
0A8C:D0 06      177          BNE  SKIP2     ; IF NO, CHECK FOR PARENS
0A8E:A5 DF      178          LDA  PARENCOUNT ; GOT COMMA, ANY PARENS?
0A90:F0 15      179          BEQ  SKIPPED   ; IF NOT, DONE
0A92:D0 ED      180          BNE  SKIP1     ; SUBSCRIPT COMMA, NEXT CHR
0A94:C9 28      181 SKIP2    CMP  #LPAREN   ; LEFT PAREN?
0A96:D0 04      182          BNE  SKIP3     ; IF NO, CHECK FOR RIGHT
0A98:E6 DF      183          INC  PARENCOUNT ; IF YES, ADD IT IN
0A9A:D0 E5      184          BNE  SKIP1     ; NEXT CHR, BRANCH ALWAYS
0A9C:C9 29      185 SKIP3    CMP  #RPAREN   ; ENDING PAREN?
0A9E:D0 E1      186          BNE  SKIP1     ; NO, JUST A CHR, GET NEXT
0AA0:C6 DF      187          DEC  PARENCOUNT ; YES, SUBTRACT IT
0AA2:10 DD      188          BPL  SKIP1     ; GET NEXT CHR
0AA4:4C C9 DE   189          JMP  SYNERR    ; CRASH ON EXTRA LPAREN
0AA7:20 98 D9   190 SKIPPED JSR  ADDON     ; UPDATE TXTPTR
0AAA:18         191          CLC            ; FORCE LEAVE WITH CARRY CLEAR
0AAB:60         192          RTS
0AAC:20 C3 09   193 PUTCOLON JSR SAVETXT   ; PUT A COLON
0AAF:20 7D 0A   194          JSR  SKIPVAR   ; AFTER THIS VARIABLE
0AB2:A5 B8      195          LDA  TXTPTR    ; SAVE ITS PLACE
0AB4:85 FA      196          STA  GENPTR    ; HERE FOR LATER
0AB6:A5 B9      197          LDA  TXTPTR+1
0AB8:85 FB      198          STA  GENPTR+1
0ABA:A0 00      199          LDY  #0        ; FIND OUT WHAT WAS THERE
0ABC:B1 B8      200          LDA  (TXTPTR),Y
0ABE:8D 19 08   201          STA  HOLDCOMMA ; SAVE IT
0AC1:F0 04      202          BEQ  COLONPUT  ; LEAVE ALONE IF EOL
0AC3:A9 3A      203          LDA  #COLON
0AC5:91 B8      204          STA  (TXTPTR),Y ; REPLACE THE COMMA (OR COLON)
0AC7:4C 97 DB   205 COLONPUT JMP  GETTXT    ; RECOVER TXTPTR AND RTS
0ACA:A0 00      206 REPCOLON LDY  #0        ; UNDO PUTCOLON
0ACC:AD 19 08   207          LDA  HOLDCOMMA ; RECOVER COMMA, COLON, OR EOL
0ACF:91 FA      208          STA  (GENPTR),Y ; PUT IT BACK
0AD1:60         209          RTS
0AD2:20 CC 09   210 ADVANCEPTRS JSR POINTDEF ; MOVE THE DEF
0AD5:20 7D 0A   211          JSR  SKIPVAR   ; DEF AND CALL PTRS
0AD8:20 DE 09   212          JSR  TXTTODEF  ; PAST THE CURRENT
0ADB:20 D5 09   213          JSR  POINTCALL ; ENTRY IN EACH LIST
0ADE:20 7D 0A   214          JSR  SKIPVAR   ; LEAVES WITH CARRY CLEAR
0AE1:4C E7 09   215          JMP  TXTTOCALL ; RTS FROM THERE
0AE4:20 C3 09   216 GETVARNAM JSR SAVETXT   ; EXIT WITH THIS TXTPTR
0AE7:A2 00      217          LDX  #0        ; MIMIC START OF PTRGET
0AE9:86 10      218          STX  DIMFLAG   ; GET THE VARIABLE NAME
0AEB:86 11      219          STX  VALTYP    ; WITHOUT GETTING THE VARIABLE
0AED:86 12      220          STX  INTFLAG   ; POINTER, AND THUS
0AEF:86 14      221          STX  SUBFLAG   ; WITHOUT MAKING A NEW
0AF1:86 DD      222          STX  ARYFLAG   ; ONE IF DOESN'T EXIST
0AF3:86 DE      223          STX  EXPRFLAG  ; ALSO FLAG EXPRS AND ARRAYS
0AF5:20 B1 00   224          JSR  CHRGET    ; TXTPTR STARTS AT LEADING COMMA
0AF8:20 7D E0   225          JSR  ISLETC    ; A LETTER?
0AFB:90 4C      226          BCC  GVEXPRSN  ; NO, LEAVE
0AFD:85 81      227          STA  LASTVAR   ; YES, SAVE 1ST CHR
0AFF:20 B1 00   228          JSR  CHRGET    ; GET SECOND
0B02:90 05      229          BCC  NAME1     ; IF NUMBER, SAVE IT
0B04:20 7D E0   230          JSR  ISLETC    ; LETTER?
0B07:90 0B      231          BCC  NAME3     ; NO, CHECK IF STRING, ETC
0B09:AA         232 NAME1    TAX            ; SAVE 2ND CHR OF NAME
0B0A:20 B1 00   233 NAME2    JSR  CHRGET    ; SKIP REST OF LETTERS
0B0D:90 FB      234          BCC  NAME2     ; AND NUMBERS. ALL EXCESS
0B0F:20 7D E0   235          JSR  ISLETC    ; CHRS IN VAR NAME
0B12:B0 F6      236          BCS  NAME2     ; SET IF LETTER
0B14:C9 24      237 NAME3    CMP  #STRING   ; GOT A "$"?
0B16:D0 06      238          BNE  NAME4     ; IF NO, CHECK %
0B18:A9 FF      239          LDA  #$FF      ; YES, FLAG IT
0B1A:85 11      240          STA  VALTYP
0B1C:D0 0C      241          BNE  NAME5     ; ALWAYS TAKEN
0B1E:C9 25      242 NAME4    CMP  #PERCENT  ; GOT AN INTEGER?
0B20:D0 0F      243          BNE  NAME6     ; NO, MAYBE PAREN
0B22:A9 80      244          LDA  #$80      ; YES, FLAG IT
0B24:85 12      245          STA  INTFLAG
0B26:05 81      246          ORA  LASTVAR   ; AND CHANGE VARNAME
0B28:85 81      247          STA  LASTVAR   ; TO REFLECT IT
0B2A:8A         248 NAME5    TXA            ; NOW SET HIGH BIT OF
0B2B:09 80      249          ORA  #$80      ; 2ND CHR TO REFLECT
0B2D:AA         250          TAX            ; INTEGER OR STRING
0B2E:           251 ;     NOTE ERROR, APPLESOFT REF MANUAL PAGE 137
0B2E:           252 ;     STRINGS ARE + ON 1ST BYTE, - ON SECOND
```

LISTING 1: SUBR.MASTER *(continued)*

```
0B2E:                253 ;       NOT THE REVERSE AS STATED THERE.
0B2E:20 B1 00        254           JSR  CHRGET     ; GET CHR AFTER % OR $
0B31:86 82           255 NAME6     STX  LASTVAR+1  ; SAVE 2ND CHR OF NAME
0B33:C9 28           256           CMP  #LPAREN    ; GOT AN ARRAY?
0B35:D0 06           257           BNE  NAME7      ; NO, CHECK IF EXPRSN
0B37:A9 FF           258           LDA  #$FF       ; ARRAY, SO FLAG IT
0B39:85 DD           259           STA  ARYFLAG
0B3B:D0 10           260           BNE  GOTVNAME   ; AND DONE
0B3D:C9 2C           261 NAME7     CMP  #COMMA     ; END OF VAR?
0B3F:F0 0C           262           BEQ  GOTVNAME   ; YES, DONE
0B41:C9 00           263           CMP  #0         ; END OF LINE?
0B43:F0 08           264           BEQ  GOTVNAME
0B45:C9 3A           265           CMP  #COLON
0B47:F0 04           266           BEQ  GOTVNAME
0B49:A9 FF           267 GVEXPRSN LDA  #$FF       ; GOT EXPRESSION
0B4B:85 DE           268           STA  EXPRFLAG
0B4D:4C 97 DB        269 GOTVNAME JMP  GETTXT     ; RECOVER TXTPTR & RTS
0B50:A6 6B           270 FINDARY LDX   ARYTAB     ; CHECK IF ARRAY
0B52:A5 6C           271           LDA  ARYTAB+1   ; OF NAME IN LASTVAR
0B54:86 9B           272 FINDA1    STX  LOWTR      ; EXISTS IN MEMORY
0B56:85 9C           273           STA  LOWTR+1    ; IF SO, POINT TO IT
0B58:C5 6E           274           CMP  STREND+1   ; IN LOWTR
0B5A:90 07           275           BCC  FINDA2     ; KEEP SEARCHING IF
0B5C:E4 6D           276           CPX  STREND     ; HAVEN'T PASSED END
0B5E:90 03           277           BCC  FINDA2     ; OF MEMORY
0B60:A0 00           278           LDY  #0         ; FLAG NO ARRAY
0B62:60              279 FINDRTS  RTS
0B63:A0 00           280 FINDA2    LDY  #0
0B65:B1 9B           281           LDA  (LOWTR),Y  ; NAME HERE MATCH
0B67:C8              282           INY             ; Y=1
0B68:C5 81           283           CMP  LASTVAR    ; ARRAY NAME?
0B6A:D0 06           284           BNE  FINDA3     ; IF NO, NEXT ARRAY
0B6C:B1 9B           285           LDA  (LOWTR),Y  ; CHECK 2ND CHR
0B6E:C5 82           286           CMP  LASTVAR+1  ; OF NAMES
0B70:F0 F0           287           BEQ  FINDRTS    ; LEAVE ON MATCH
0B72:C8              288 FINDA3    INY             ; (Y=2) GET OFFSET TO NEXT
0B73:B1 9B           289           LDA  (LOWTR),Y  ; ARRAY
0B75:18              290           CLC             ; ADD IT TO LOWTR
0B76:65 9B           291           ADC  LOWTR      ; TO POINT TO NEXT
0B78:AA              292           TAX             ; GOT LOW BYTE
0B79:C8              293           INY             ; (Y=3) GET HIGH BYTE
0B7A:B1 9B           294           LDA  (LOWTR),Y
0B7C:65 9C           295           ADC  LOWTR+1
0B7E:90 D4           296           BCC  FINDA1     ; BRANCH ALWAYS TAKEN
0B80:20 E4 0A        297 MAKEVAR   JSR  GETVARNAM  ; WHAT TYPE OF VAR?
0B83:A5 DD           298           LDA  ARYFLAG    ; IF ARRAY
0B85:D0 35           299           BNE  MAKEARRAY  ; DO BELOW
0B87:A5 DE           300           LDA  EXPRFLAG   ; IF EXPRESSION
0B89:F0 03           301           BEQ  MAKESIMPLE ; THEN
0B8B:4C C9 DE        302           JMP  SYNERR     ; CRASH
0B8E:20 38 0A        303 MAKESIMPLE JSR SAVEARYTAB ; MAKE A LOCAL
0B91:A5 69           304           LDA  VARTAB     ; SIMPLE BY HIDING
0B93:85 6B           305           STA  ARYTAB     ; ALL THE OLD ONES
0B95:A5 6A           306           LDA  VARTAB+1   ; THUS CREATING A
0B97:85 6C           307           STA  ARYTAB+1   ; NEW ONE AT THE
0B99:20 B1 00        308           JSR  CHRGET     ; BOTTOM OF STORAGE. PASS
0B9C:20 E3 DF        309           JSR  PTRGET     ; LEADING COMMA.
0B9F:AD 0F 08        310           LDA  OLDARYTAB  ; RECOVER ARRAYS
0BA2:18              311           CLC             ; TO REFLECT NEW
0BA3:69 07           312           ADC  #7         ; SIMPLE IN MEMORY.
0BA5:85 6B           313           STA  ARYTAB     ; UPDATE ARRAY PTR
0BA7:AD 10 08        314           LDA  OLDARYTAB+1
0BAA:69 00           315           ADC  #0         ; ADD IN CARRY
0BAC:85 6C           316           STA  ARYTAB+1
0BAE:AD 15 08        317           LDA  OLDSIMPLE  ; UPDATE POINTER
0BB1:69 07           318           ADC  #7         ; (CARRY CLEAR) TO START
0BB3:8D 15 08        319           STA  OLDSIMPLE  ; OF ORIGINAL
0BB6:90 03           320           BCC  SIMPLERTS  ; SIMPLE VARIABLES
0BB8:EE 16 08        321           INC  OLDSIMPLE+1
0BBB:60              322 SIMPLERTS RTS
0BBC:20 AC 0A        323 MAKEARRAY JSR PUTCOLON   ; ONLY DIM THIS ARRAY
0BBF:20 50 0B        324           JSR  FINDARY    ; NOW DOES THIS ARRAY EXIST?
0BC2:98              325           TYA             ; IF SO, Y NOT 0. IF EXISTS,
0BC3:D0 09           326           BNE  MAKEAR1    ; MAKE NEW ARRAY, SAME NAME
0BC5:20 B1 00        327           JSR  CHRGET     ; NO ARRAY OF THIS NAME EXISTS
0BC8:20 D9 DF        328           JSR  DIM        ; MAKE ONE NOW
0BCB:4C CA 0A        329           JMP  REPCOLON   ; FIX LINE AND LEAVE
0BCE:20 38 0A        330 MAKEAR1   JSR  SAVEARYTAB ; ARRAY WITH THIS NAME
0BD1:20 4E 0A        331           JSR  SAVESTREND ; EXISTS. TO MAKE A NEW ONE,
0BD4:85 6C           332           STA  ARYTAB+1   ; HIDE ALL OLD ARRAYS BY
0BD6:A5 6D           333           LDA  STREND     ; CALLING THEM SIMPLE VARS.
0BD8:85 6B           334           STA  ARYTAB     ; THEN MAKE NEW ARRAY AT THE
0BDA:20 B1 00        335           JSR  CHRGET     ; (NEEDED FOR DIM)
0BDD:20 D9 DF        336           JSR  DIM        ; TOP OF MEMORY.
0BE0:20 97 DB        337           JSR  GETTXT     ; RECOVER TXTPTR
0BE3:AD 0F 08        338           LDA  OLDARYTAB  ; SET UP BLTU
0BE6:85 9B           339           STA  LOWTR      ; TO TRANSFER ARRAYS.
0BE8:AD 10 08        340           LDA  OLDARYTAB+1 ; TRANSFER THE OLD ARRAYS
0BEB:85 9C           341           STA  LOWTR+1    ; FROM ARYTAB THROUGH OLDSTREND
0BED:AD 13 08        342           LDA  OLDSTREND  ; UP IN MEMORY ENDING AT THE
0BF0:18              343           CLC             ; NEW STREND, OVERWRITING THE
0BF1:69 01           344           ADC  #1         ; NEW ARRAY.
0BF3:85 96           345           STA  HIGHTR     ; USE BLTUP NOT BLTU
0BF5:AD 14 08        346           LDA  OLDSTREND+1 ; TO DO THE MOVE AS WE
0BF8:69 00           347           ADC  #0         ; KNOW STREND IS OK (NO NEED FOR
0BFA:85 97           348           STA  HIGHTR+1   ; 'REASON' ROUTINE) AND DON'T
```

the array, but it can take longer. If another array with the DEF name already exists and is located lower in memory than the CALL array to be renamed, the CALL array is moved below it. No variable space is used when renaming or moving the arrays.

Integer and real array data takes different amounts of memory. If array R() is renamed as I%() and you reference I%(30), you will not get the value of R(30). We considered adding the code needed to define a local R() for the subroutine, to convert I%()'s values to real, and to pass them to R(), but this would be slow and would waste the space taken up by R(). We were concerned that automating the practice would encourage it, and decided not to. Instead, the program flags an attempted pass of this sort with a TYPE MISMATCH error.

The CALL array's subscripts must be legal. If the subscript is too large, given the DIMension of the array, or if there are too many dimensions, the program halts with the usual BAD SUBSCRIPT message. Otherwise, the values of the subscripts are immaterial. The whole array is passed to the DEF list array. The subscripts of the DEF list arrays are not checked in any way. Subscripts are not even necessary — D() will do the job in the DEF statement. The parentheses specify that we're dealing with an array, and the D specifies that it is a real array named D. The dimension assigned to D() is the actual dimension of the CALL list array. If the subroutine tries to use D() with a bad subscript for the passed CALL array, BASIC will flag this.

The requirement that the CALL array must have been dimensioned uses extra code, but it adds protection against various types of errors. If the CALL array was not DIMensioned before our handler tries to pass it to the DEF array, the program halts with an ARRAY ERROR message. Our main concern in adding this reflects our feeling that arrays should always be DIMensioned explicitly. Traditional BASIC does the programmer a "favor" in allowing various types of sloppy coding practices, including this one. We'd rather be spared the favor, and the errors we've missed finding because of it.

Preferences aside, this provides protection against nesting problems that can arise if you accidentally have one too many NEXTs or RETURNs in your subroutine. If there is an active FOR or GOSUB outside the subroutine, one of these may pull you back to it, without an error message, but also without properly exiting the called subroutine. (We'll describe the problem in more detail below.) Here is the basic idea: suppose that you do somehow get back to the calling program without exiting from this subroutine. If you passed an array down, it is renamed for the subroutine. Without the EXIT, it is not renamed back. The next time you call the routine, no array with the name in the CALL statement will exist, forcing a program halt with an error message. This tech-

LISTING 1: SUBR.MASTER *(continued)*

```
0BFC:A5 6D      349          LDA  STREND     ; WANT ANY PTRS TINKERED WITH.
0BFE:69 01      350          ADC  #1
0C00:85 94      351          STA  HIGHDS
0C02:20 4E 0A   352          JSR  SAVESTREND ; RETURNS HOLDING STREND+1
0C05:69 00      353          ADC  #0         ; WITH CARRY UNAFFECTED
0C07:85 95      354          STA  HIGHDS+1
0C09:20 9A D3   355          JSR  BLTUP      ; ACTUAL MOVE HERE.
0C0C:20 43 0A   356          JSR  GETARYTAB  ; NOW HIDE THE ARRAYS AGAIN
0C0F:85 6E      357          STA  STREND+1   ; (=ARYTAB+1). THIS TIME THE
0C11:A5 6B      358          LDA  ARYTAB     ; DIM CREATES THE ARRAY AT THE
0C13:85 6D      359          STA  STREND     ; BOTTOM OF ARRAY STORAGE, SO
0C15:20 B1 00   360          JSR  CHRGET     ; IT IS ALWAYS FOUND FIRST.
0C18:20 D9 DF   361          JSR  DIM        ; THE MOVE ABOVE MADE ROOM HERE
0C1B:20 59 0A   362          JSR  GETSTREND  ; FOR THIS.  RECOVER CORRECT
0C1E:20 CA 0A   363          JSR  REPCOLON   ; STREND, FIX LINE,
0C21:60         364 RTS1     RTS             ; AND DONE.
0C22:A2 02      365 RENAME   LDX  #2         ; PUT NEW NAMES IN OLD ARRAYS,
0C24:E4 DA      366 RENAME1  CPX  BUFPTR     ; WHILE ARRAYS LEFT TO DO.
0C26:B0 F9      367          BCS  RTS1       ; THEN EXIT
0C28:BD 1A 08   368          LDA  SECBUF,X   ; LOCATION OLD ARRAY
0C2B:85 9B      369          STA  LOWTR      ; STASH IT
0C2D:BD 1B 08   370          LDA  SECBUF+1,X
0C30:85 9C      371          STA  LOWTR+1
0C32:A0 00      372          LDY  #0         ; INDIRECT ADDRESSING
0C34:BD 18 08   373          LDA  SECBUF-2,X ; NEW NAME
0C37:91 9B      374          STA  (LOWTR),Y  ; FOR THE ARRAY
0C39:BD 19 08   375          LDA  SECBUF-1,X
0C3C:C8         376          INY
0C3D:91 9B      377          STA  (LOWTR),Y
0C3F:8A         378          TXA
0C40:69 04      379          ADC  #4         ; CLEAR CARRY FROM BCS ABOVE
0C42:AA         380          TAX
0C43:90 DF      381          BCC  RENAME1    ; MUST STILL BE CLEAR
0C45:           382 **********************************
0C45:           383 *        CALL THIS ADDRESS         *
0C45:           384 *        TO ENTER A PROC           *
0C45:           385 **********************************
0C45:20 AA 09   386 PROC     JSR  IN         ; CHECK MODE. SAVE FA-FF.
0C48:A5 B8      387          LDA  TXTPTR     ; CALL VARLIST STARTS
0C4A:8D 09 08   388          STA  CALLIST    ; AT COMMA (OR EOL IF NO LIST)
0C4D:A5 B9      389          LDA  TXTPTR+1   ; AFTER PROC NAME, WHICH
0C4F:8D 0A 08   390          STA  CALLIST+1  ; IS WHERE TXTPTR IS AT.
0C52:A0 00      391          LDY  #0         ; FOR INDIRECT ADDRESS
0C54:20 19 0A   392 FINDNAME JSR  DECTXT     ; MOVE TXTPTR BACK
0C57:B1 B8      393          LDA  (TXTPTR),Y ; TO FIND "CALL"
0C59:C9 8C      394          CMP  #CALL      ; NAME OF THE PROC
0C5B:D0 F7      395          BNE  FINDNAME   ; STARTS THERE.
0C5D:20 E7 09   396          JSR  TXTTOCALL  ; POINT TO IT IN CALLPTR
0C60:8D 0E 08   397          STA  PROCNAME+1 ; AND IN PTR TO
0C63:A5 B8      398          LDA  TXTPTR     ; THE NAME ITSELF.
0C65:8D 0D 08   399          STA  PROCNAME
0C68:20 97 D6   400          JSR  STXTPT     ; TXTPTR AT START OF PROG
0C6B:A0 02      401 FINDDEF  LDY  #2         ; HIGH BYTE OF
0C6D:B1 B8      402          LDA  (TXTPTR),Y ; NEXT LINE'S ADDRESS
0C6F:D0 05      403          BNE  FINDDEF1   ; IF 0, NO PROG LEFT
0C71:A2 5A      404 UNDEF    LDX  #90        ; IN WHICH CASE MAKE
0C73:4C 12 D4   405          JMP  ERROR      ; UNDEFINED STATEMENT ERR.
0C76:C8         406 FINDDEF1 INY             ; Y=3. LOW BYTE OF
0C77:B1 B8      407          LDA  (TXTPTR),Y ; NEW LINE #
0C79:8D 0B 08   408          STA  DEFLINE    ; SAVE IT
0C7C:C8         409          INY             ; Y=4
0C7D:B1 B8      410          LDA  (TXTPTR),Y ; HIGH BYTE
0C7F:8D 0C 08   411          STA  DEFLINE+1  ; GOT IT
0C82:C8         412          INY             ; FIRST CHR OF TEXT
0C83:B1 B8      413          LDA  (TXTPTR),Y ; IS IT "DEF"?
0C85:C9 B8      414          CMP  #DEF
0C87:D0 16      415          BNE  NEXTLINE   ; IF NOT, TRY NEXT LINE
0C89:20 98 D9   416          JSR  ADDON      ; POINT TO IT WITH TXTPTR
0C8C:A0 00      417          LDY  #0         ; Y INDEXES DEF AND CALL NAMES
0C8E:C8         418 NEXTCHAR INY             ; PAST "DEF" & "CALL"
0C8F:B1 FC      419          LDA  (CALLPTR),Y ; GET NEXT CHAR OF NAME
0C91:F0 17      420          BEQ  CNAMDONE   ; CALL NAME ENDS ON 0
0C93:C9 3A      421          CMP  #COLON     ; OR ":" OR ","
0C95:F0 13      422          BEQ  CNAMDONE   ; IF END, CHECK IF
0C97:C9 2C      423          CMP  #COMMA     ; DEF NAME DONE TOO.
0C99:F0 0F      424          BEQ  CNAMDONE   ; IF GET PAST HERE, THEN
0C9B:D1 B8      425          CMP  (TXTPTR),Y ; STILL IN NAME. CHECK DEF.
0C9D:F0 EF      426          BEQ  NEXTCHAR   ; BOTH MATCH, CHECK NEXT CHR
0C9F:20 98 D9   427 NEXTLINE JSR  ADDON      ; DEF & CALL MISMATCH. MOVE
0CA2:20 A6 D9   428          JSR  REMN       ; PAST LINE #, THEN PAST LINE
0CA5:20 98 D9   429          JSR  ADDON      ; SET TXTPTR TO NEXT LINE
0CA8:D0 C1      430          BNE  FINDDEF    ; ALWAYS TAKEN.  TRY AGAIN
0CAA:D1 B8      431 CNAMDONE CMP  (TXTPTR),Y ; DEF NAME DONE TOO?
0CAC:F0 0C      432          BEQ  FOUNDIT    ; YES, DONE SEARCHING
0CAE:C9 2C      433          CMP  #COMMA     ; IF COMMA & DEF NOT COMMA
0CB0:F0 ED      434          BEQ  NEXTLINE   ; MISMATCH OR BAD CALL
0CB2:B1 B8      435          LDA  (TXTPTR),Y ; CALL IS END OF LINE
0CB4:F0 04      436          BEQ  FOUNDIT    ; MATCH IF DEF IS TOO
0CB6:C9 3A      437          CMP  #COLON
0CB8:D0 E5      438          BNE  NEXTLINE
0CBA:20 98 D9   439 FOUNDIT  JSR  ADDON      ; TXTPTR POINTS TO COMMA
0CBD:A5 B8      440          LDA  TXTPTR     ; AT START OF DEF LIST
0CBF:8D 07 08   441          STA  DEFLIST    ; SAVE THE POINTER
0CC2:A5 B9      442          LDA  TXTPTR+1   ; IN PTR TO START OF
0CC4:8D 08 08   443          STA  DEFLIST+1  ; DEF VAR LIST
0CC7:20 F0 09   444          JSR  STARTLIST  ; MOVE PTRS TO
```

nique will not always catch bad nesting, but it is another level of safeguarding.

Nesting

You can "nest" subroutines to your heart's content or until the Apple runs out of memory, whichever comes first. If one subroutine calls a second, the second is "nested" within the first. If the second calls a third, you've added another level of nesting. Each level of nesting has a memory cost associated with it, which disappears on exit. There is a basic cost of 21 bytes per level of nesting (for pointers, etc.), plus seven bytes per simple variable passed, plus however much is required for variables declared LOCAL. Note that locals from a calling subroutine are globals to the called one, as in Pascal.

Speed

Calling and exiting subroutines takes a variable length of time, depending on how many variables are passed, how many locals are created, and how many variables are already in memory. The dominant factor is the number of bytes taken up by simple and array storage. You can estimate how long a CALL or EXIT will take by determining how many bytes are taken up by variable storage (subtract VARTAB, in $69,$6A, from STREND, in $6D,$6E, for this) and how many local simple variables are created during the CALL. Each byte takes about 17.5 microseconds to move, and each is moved whenever a simple variable is created or cleared. A minimum of three simple variables are created per CALL and cleared per EXIT, for housekeeping. The move is a bit faster when lots of data is transferred, and a bit slower per byte when very few bytes are moved, but this is a good ballpark figure, even though it ignores local array handling. In practical terms, if there is very little data, a CALL–EXIT pair takes about 0.2 seconds to execute. If memory is nearly full (say, 25,000 bytes of data), each CALL–EXIT pair takes a minimum of 2.5 seconds.

ERROR HANDLING

Detected Errors

Whenever possible, we rely on Applesoft to detect errors, either when our handler uses Applesoft internal routines, or within the subroutine itself. For example, we don't check if a DEF array has the right number of dimensions. If it has the wrong number for the CALL array being passed to it, Applesoft will halt the program as soon as that DEF array is used in the subroutine.

A number of further errors, which we have to catch, can arise relating to our subroutines themselves. For these we either use standard Applesoft error messages or, in two cases, parts of them: ARRAY ERROR and MEMORY ERROR. If you get either of these, you know the program crashed while executing a CALL or an EXIT.

An ARRAY ERROR occurs in response

to various errors involving array passing. For example, if a simple variable is passed to a DEF array, if an attempt to pass a previously undefined array to a subroutine is detected, or if a DEF array is part of an expression, you get an ARRAY ERROR message.

A MEMORY ERROR occurs when the pointer to the end of variable storage (STREND) doesn't have the value at a certain point during the EXIT that it had at a comparable point during the CALL. Usually this means that you created a new global or cleared an old one within the subroutine. It also signals crossed subroutines and, generally, an EXIT from a subroutine with a different name from the one called.

An UNDEF'D STATEMENT error indicates that the handler can't find a subroutine of the name you called. This happens most frequently when you miss a comma after the subroutine name following the CALL or DEF. CALL SRTX(0),10 will not lead you to DEF SRT,S(0),N.

CALL and DEF parameter lists that have a different number of items may be flagged in a number of ways. If the CALL or the DEF statement is followed by no list (and no comma), while the other is followed by a parameter list, you get an UNDEF'D STATEMENT error. If both statements have a parameter list but one list has more items than the other, you get a SYNTAX ERROR, either on entry to the routine (CALL list short) or on exit (DEF list short) — if you don't get a MEMORY ERROR first.

Undetected Errors

Some errors are not trapped, since we consider the cure worse than the disease. Most of these errors are very unlikely to occur, or generally harmless. Further, many of them are eventually caught when BASIC or the handler doesn't understand something later in the program. However, we'll describe these errors as if they are never caught, and consider the "worst case" behavior of the program. Our intention is to clearly discuss the error handling problems you may run into, and what to do about them if you do. (Don't be scared off, though — in practice, we've found this program to be extremely reliable.)

Expressions in the DEF Variable List — If you have an expression in the DEF list that starts with an array and ends with a parenthesis (but didn't start with a parenthesis), it must start with the array name itself. If the corresponding CALL variable is also an array, then and only then, the handler will not crash on an expression in the list. Instead it will pass the CALL array to the DEF array and back to EXIT, completely ignoring the expression code between the array name and the next comma.

LISTING 1: SUBR.MASTER *(continued)*

```
00CA:A5 69     445          LDA  VARTAB     ; START OF VAR LISTS.
00CC:8D 15 08  446          STA  OLDSIMPLE  ; THEN SAVE PTR TO
00CF:A5 6A     447          LDA  VARTAB+1   ; START OF ENTERING
00D1:8D 16 08  448          STA  OLDSIMPLE+1 ; SIMPLE VARIABLES
00D4:20 CC 09  449 PASSIMPLE JSR POINTDEF  ; PNT TO NEXT DEF LIST VAR
00D7:20 B7 00  450          JSR  CHRGOT     ; WHILE NOT EOL PASS SIMPLES.
00DA:F0 30     451          BEQ  PASSARY    ; WHEN DONE, PASS THE ARRAYS
00DC:20 E4 0A  452          JSR  GETVARNAM  ; SIMPLE VAR?
00DF:A5 DE     453          LDA  EXPRFLAG   ; NO EXPRS IN DEFS
00E1:F0 06     454          BEQ  SIMPLE1    ; CHECK IF ARRAY BELOW
00E3:20 72 0A  455          JSR  DEFTOCUR   ; CRASH ON EXPR
00E6:4C C9 DE  456          JMP  SYNERR     ; IN DEF STATEMENT
00E9:A5 DD     457 SIMPLE1  LDA  ARYFLAG    ; ARRAY?
00EB:D0 1A     458          BNE  NEXTSIMPLE ; IF SO, SKIP IT
00ED:20 8E 0B  459          JSR  MAKESIMPLE ; CREATE LOCAL WITH
00F0:20 D5 09  460          JSR  POINTCALL  ; FIND VAR IN CALL LIST
00F3:20 B1 00  461          JSR  CHRGET     ; MOVE PAST LEADING COMMA
00F6:20 64 0A  462          JSR  SIMPTOVAR  ; LOOK PAST LOCAL SIMPLES
00F9:A5 83     463          LDA  VARPNT     ; SET UP THESE PTRS FOR "LET"
00FB:85 85     464          STA  FORPNT     ; THEY WERE SET UP BY THE PTRGET
00FD:A5 84     465          LDA  VARPNT+1   ; CALL IN MAKESIMPLE.
00FF:85 86     466          STA  FORPNT+1   ; GO PARTWAY INTO LET TO
0101:20 52 DA  467          JSR  LETCNT     ; SKIP THE "=" TEST THERE.
0104:20 2D 0A  468          JSR  GETVARTAB  ; RECOVER TRUE VARTAB
0107:20 D2 0A  469 NEXTSIMPLE JSR ADVANCEPTRS ; UPDATE CALL & DEF PTRS
010A:90 C8     470          BCC  PASSIMPLE  ; ALWAYS CLEAR
010C:20 F0 09  471 PASSARY  JSR  STARTLIST  ; RECOVER LIST PTRS
010F:A9 00     472          LDA  #0         ; INITIALIZE COUNTERS
0111:85 DA     473          STA  BUFPTR
0113:85 DB     474          STA  COUNTER
0115:20 CC 09  475 ARRAY1   JSR  POINTDEF   ; GET 1ST VAR IN DEF
0118:20 B7 00  476          JSR  CHRGOT     ; WHILE NOT EOL, PASS ARRAYS
011B:D0 03     477          BNE  ARRAY2
011D:4C AF 0D  478          JMP  FIND       ; WHEN DONE, RENAME & FIND THEM
0120:20 E4 0A  479 ARRAY2   JSR  GETVARNAM  ; SIMPLE VAR?
0123:A5 DD     480          LDA  ARYFLAG    ; NOT IF THIS NOT ZERO
0125:D0 05     481          BNE  ARRAY3     ; SO PASS IT
0127:20 D2 0A  482          JSR  ADVANCEPTRS ; SIMPLE, SO SKIP IT
012A:90 E9     483          BCC  ARRAY1     ; LOOK AT NEXT VAR
012C:A5 81     484 ARRAY3   LDA  LASTVAR    ; STORE NAME
012E:A6 DA     485          LDX  BUFPTR     ; IN SECBUF
0130:9D 1A 08  486          STA  SECBUF,X
0133:29 80     487          AND  #$80       ; STORE TYPE
0135:9D 1C 08  488          STA  SECBUF+2,X ; HERE TEMPORARILY
0138:A5 82     489          LDA  LASTVAR+1  ; NAME HIGH BYTE
013A:9D 1B 08  490          STA  SECBUF+1,X
013D:29 80     491          AND  #$80
013F:9D 1D 08  492          STA  SECBUF+3,X
0142:20 D5 09  493          JSR  POINTCALL  ; FIND CALL ARRAY
0145:20 E4 0A  494          JSR  GETVARNAM  ; GOT ARRAY?
0148:A5 DD     495          LDA  ARYFLAG    ; MUST BE FF
014A:D0 05     496          BNE  ARYCHK     ; OR CRASH WITH
014C:A2 80     497 ARRAYERR LDX #128        ; "ARRAY ERROR"
014E:4C 12 D4  498          JMP  ERROR      ; AS PASSING SIMPLE TO ARRAY
0151:20 50 0B  499 ARYCHK   JSR  FINDARY    ; WHERE IS IT?
0154:98        500          TYA             ; REQUIRE THAT IT EXISTS ALREADY
0155:F0 F5     501          BEQ  ARRAYERR   ; REFUSE TO PASS UNDIMENSIONED ARRAY.
0157:20 D5 09  502          JSR  POINTCALL  ; NOW SEE IF ARRAY EXPRESSION
015A:20 B1 00  503          JSR  CHRGET     ; MUST HAVE COMMA OR EOL AFTER
015D:20 E3 DF  504          JSR  PTRGET     ; ARRAY NAME.
0160:20 B7 00  505          JSR  CHRGOT     ; DO WE?
0163:F0 04     506          BEQ  ITISARRAY
0165:C9 2C     507          CMP  #COMMA
0167:D0 E3     508          BNE  ARRAYERR
0169:20 D2 0A  509 ITISARRAY JSR ADVANCEPTRS ; UPDATE THE LIST PTRS
016C:20 CC 09  510          JSR  POINTDEF   ; NOW CHECK THAT DEF ARRAY
016F:20 19 0A  511          JSR  DECTXT     ; WAS NOT AN EXPRESSION WITH
0172:20 B7 00  512          JSR  CHRGOT     ; A LEADING ARRAY IN IT
0175:C9 29     513          CMP  #RPAREN    ; SUCH AS D(1)+5
0177:F0 06     514          BEQ  BOTHOK     ; THIS CHECKS THAT LAST CHAR
0179:20 72 0A  515          JSR  DEFTOCUR   ; OF DEF VAR IS ')'.
017C:4C 4C 0D  516          JMP  ARRAYERR
017F:A6 DA     517 BOTHOK   LDX  BUFPTR     ; WHERE WERE WE?
0181:A5 81     518          LDA  LASTVAR    ; COMPARE TYPES OF VARS
0183:29 80     519          AND  #$80       ; MASK ALL BUT TYPE FLAG
0185:DD 1C 08  520          CMP  SECBUF+2,X ; STORED HERE FOR DEF VAR
0188:D0 0E     521          BNE  BADTYPE    ; CRASH IF NOT SAME
018A:A5 9B     522          LDA  LOWTR      ; OVERWRITE TYPE WITH
018C:9D 1C 08  523          STA  SECBUF+2,X ; ADDRESS OF CALL VAR
018F:A5 82     524          LDA  LASTVAR+1  ; DO SAME FOR HIGH BYTE
0191:29 80     525          AND  #$80
0193:DD 1D 08  526          CMP  SECBUF+3,X
0196:F0 03     527          BEQ  ARRAY4
0198:4C 76 DD  528 BADTYPE  JMP  MISMATCH
019B:A5 9C     529 ARRAY4   LDA  LOWTR+1
019D:9D 1D 08  530          STA  SECBUF+3,X
01A0:8A        531          TXA             ; UPDATE BUFPTR
01A1:69 03     532          ADC  #3         ; CARRY SET, THIS ADDS 4
01A3:85 DA     533          STA  BUFPTR
01A5:C9 D0     534          CMP  #BUFMAX    ; PAST THE MAX # VARS?
01A7:B0 03     535          BCS  TOOMANY    ; IF SO, CRASH OUT OF MEMORY
01A9:4C 15 0D  536          JMP  ARRAY1     ; ELSE, DO NEXT ARRAY
01AC:4C 10 D4  537 TOOMANY  JMP  OMERR
01AF:20 22 0C  538 FIND     JSR  RENAME     ; RENAME THE ARRAYS
01B2:A2 02     539          LDX  #2         ; SEARCH FOR NEWLY RENAMED ARRAYS
01B4:E4 DA     540 FIND1    CPX  BUFPTR     ; WHILE ARRAYS LEFT TO CHECK
```

Missed Commas in the CALL and DEF Lists — If you miss the comma after the subroutine name in both lists, and if the first variable in each list is simple and both have the same name, the handler will think that that variable is part of the name and won't realize you missed the comma. Since the variable passed is an existing global, the subroutine will execute correctly. If you don't miss the comma in your next CALL, however, you may wonder why the program didn't crash on the first one.

Wrong EXIT Subroutine Name — If you CALL SRT and try to exit from STR, the program will crash, as it should. But suppose you CALL SUB2 and CALL EXIT, SUB1. In this case, since SUB1 and SUB2 are the same variable to BASIC, you will not get an error message. Instead, you will exit from SUB2 normally. This is no problem unless you compound the error. If subroutine SUB1 includes a CALL to SUB2, which in turn tries to EXIT SUB1 (lots of GOTOs in a program could put you in this position), then EXIT will behave just like a RETURN would and take you out of the last subroutine called, i.e., SUB2. Again, this can only happen if two routines have the same first two letters, and if one calls the other directly, without a third one between them.

Invalid User-Defined Functions — If you DEF FN A(X) in the main program and try to do anything with it in the subroutine, the computer will respond with the wrong answer but no error message.

Wrong Value for Exit or the Subroutine Name — If you CALL SRT, and SRT is not 3141; or CALL EXIT, SRT, and EXIT is not 4058, BASIC will transfer control to the wrong location in memory. The typical case is that you forget to define one of these variables before calling it. In this case, you CALL 0, which works like an END statement: a halt with no error message. If the variable is non-zero but wrong, you'll likely crash on an error test in our routine, or crash with a Monitor break on a stored zero. But anything is possible.

Immediate Mode GOSUB — You can only call subroutines in program execution mode because we use the keyboard input buffer at $200 to move arrays. Immediate mode GOSUBs may cause data to be scrambled without any error message. Therefore, you should only use Subroutine Master from a running program.

Crossing GOSUB or FOR With Called Subroutines — If you call a subroutine from within a GOSUB subroutine, and the called subroutine contains one too many RETURN statements, then you will return to the outer GOSUB without getting a RETURN WITHOUT GOSUB error message and

LISTING 1: SUBR.MASTER *(continued)*

```
ØDB6:BØ 38      541          BCS  SORT       ; THEN SORT THOSE TO MOVE
ØDB8:BD 18 Ø8   542          LDA  SECBUF-2,X ; GET THE NAME
ØDBB:85 81      543          STA  LASTVAR    ; PUT IT HERE
ØDBD:BD 19 Ø8   544          LDA  SECBUF-1,X ; TO FIND IT
ØDC0:85 82      545          STA  LASTVAR+1
ØDC2:86 DC      546          STX  COUNTER+1  ; AUXILIARY COUNTER
ØDC4:20 50 ØB   547          JSR  FINDARY    ; WHERE'S THE ARRAY?
ØDC7:A6 DC      548          LDX  COUNTER+1  ; RECOVER X
ØDC9:A5 9C      549          LDA  LOWTR+1    ; ARRAY FOUND STARTS HERE
ØDCB:DD 1B Ø8   550          CMP  SECBUF+1,X ; SAME AS RENAMED ONE?
ØDCE:DØ Ø7      551          BNE  MUSTMOVE   ; IF NOT, HAVE TO MOVE
ØDDØ:A5 9B      552          LDA  LOWTR      ; THE NEW ONE DOWN IN
ØDD2:DD 1A Ø8   553          CMP  SECBUF,X   ; MEMORY, TO HIDE THIS ONE.
ØDD5:FØ 12      554          BEQ  FIND3      ; SAME ARRAY, CHECK NEXT
ØDD7:A4 DB      555 MUSTMOVE LDY  COUNTER    ; INDEX BUFR
ØDD9:BD 1A Ø8   556          LDA  SECBUF,X   ; RECOVER ADDRESS OF
ØDDC:99 ØØ Ø2   557          STA  BUFR,Y     ; RENAMED ARRAY
ØDDF:BD 1B Ø8   558          LDA  SECBUF+1,X ; AND SAVE IT FOR SORTING
ØDE2:99 Ø1 Ø2   559          STA  BUFR+1,Y
ØDE5:C8         560          INY
ØDE6:C8         561          INY
ØDE7:84 DB      562          STY  COUNTER    ; POINT TO NEXT FREE SPOT
ØDE9:8A         563 FIND3    TXA             ; UPDATE BUFFER PTR
ØDEA:18         564          CLC
ØDEB:69 Ø4      565          ADC  #4
ØDED:AA         566          TAX
ØDEE:DØ C4      567          BNE  FIND1      ; ALWAYS TAKEN
ØDFØ:A4 DB      568 SORT     LDY  COUNTER    ; WHILE ARRAYS TO SORT, DO
ØDF2:DØ Ø3      569          BNE  SORT1      ; ELSE DONE PASSING.  NOTE NON-Ø
ØDF4:4C FA ØE   570          JMP  LOCAL      ; COUNTER PTS 1 PAST END OF LIST
ØDF7:           571 ;
ØDF7:           572 ;     WE WILL MOVE THE ARRAYS IN ORDER FROM
ØDF7:           573 ;     LOWEST IN MEMORY TO HIGHEST.  THIS IS FASTER
ØDF7:           574 ;     AND DOES NOT INTERFERE WITH THE ADDRESSES
ØDF7:           575 ;     STORED FOR THE VARIABLES TO BE MOVED.
ØDF7:           576 ;
ØDF7:A2 FE      577 SORT1    LDY  #$FE       ; START AT Ø AFTER INY'S
ØDF9:C8         578 NEXTY    INY             ; FOR Y=FIRST TO LAST-1 ARRAY
ØDFA:C8         579          INY             ; (ARRAY PTRS TAKE 2 BYTES)
ØDFB:C4 DB      580          CPY  COUNTER    ; WHILE ARRAYS LEFT, SORT
ØDFD:B2 41      581          BCS  SORTED     ; THEN MOVE THEM
ØDFF:98         582          TYA             ; FOR X = Y+1TH TO LAST ARRAY
ØE00:AA         583          TAX
ØE01:E8         584 NEXTX    INX             ; IF ADDRESS(Y) > ADDRESS(X)
ØE02:E8         585          INX             ; THEN SWITCH ORDER OF ADDRESSES
ØE03:E4 DB      586          CPX  COUNTER    ; ANY LEFT IN LIST?
ØE05:B2 F2      587          BCS  NEXTY      ; IF NO, DONE INSIDE LOOP
ØE07:B9 Ø1 Ø2   588          LDA  BUFR+1,Y   ; COMPARE HIGH BYTES
ØE0A:DD Ø1 Ø2   589          CMP  BUFR+1,X
ØE0D:90 F2      590          BCC  NEXTX      ; ADDRESS(Y) < ADDRESS(X)
ØE0F:D2 ØD      591          BNE  SWITCH     ; ADDRESS(Y) > ADDRESS(X)
ØE11:B9 ØØ Ø2   592          LDA  BUFR,Y     ; CHECK LOW BYTES
ØE14:DD ØØ Ø2   593          CMP  BUFR,X
ØE17:90 E8      594          BCC  NEXTX
ØE19:D2 Ø3      595          BNE  SWITCH
ØE1B:4C 4C Ø2   596          JMP  ARRAYERR   ; FLAG ALIASED ARRAYS WHEN SPOTTED.
ØE1E:B9 ØØ Ø2   597 SWITCH   LDA  BUFR,Y     ; STASH ADDRESS(Y)
ØE21:85 FA      598          STA  GENPTR     ; HERE, TEMPORARILY
ØE23:B9 Ø1 Ø2   599          LDA  BUFR+1,Y
ØE26:85 FB      600          STA  GENPTR+1
ØE28:BD ØØ Ø2   601          LDA  BUFR,X     ; NOW SET ADDRESS(Y)
ØE2B:99 ØØ Ø2   602          STA  BUFR,Y     ; TO ADDRESS(X)
ØE2E:BD Ø1 Ø2   603          LDA  BUFR+1,X
ØE31:99 Ø1 Ø2   604          STA  BUFR+1,Y
ØE34:A5 FA      605          LDA  GENPTR     ; AND SET ADDRESS(X)
ØE36:9D ØØ Ø2   606          STA  BUFR,X     ; TO OLD ADDRESS(Y)
ØE39:A5 FB      607          LDA  GENPTR+1
ØE3B:9D Ø1 Ø2   608          STA  BUFR+1,X
ØE3E:B2 C1      609          BCS  NEXTX      ; SET ENTERING SWITCH
ØE40:BD ØØ Ø2   610 SORTED   LDA  BUFR,X     ; NOW MOVE SORTED ADDRESSES
ØE43:9D 1A Ø8   611          STA  SECBUF,X   ; BACK UP TO SECBUF.  SINCE
ØE46:CA         612          DEX             ; THIS LIST IS HALF OLD SECBUF
ØE47:1Ø F7      613          BPL  SORTED     ; LENGTH, CAN'T EXCEED $7F
ØE49:E8         614          INX             ; SET X BACK TO Ø
ØE4A:86 DA      615          STX  BUFPTR     ; INITIALIZE BUFFER POINTER
ØE4C:20 38 ØA   616          JSR  SAVEARYTAB ; WILL OVERWRITE THIS
ØE4F:8D 18 Ø8   617          STA  NEWARYTAB+1 ; TO AVOID MOVING
ØE52:A5 6B      618          LDA  ARYTAB     ; MOVED ARRAYS TWICE
ØE54:8D 17 Ø8   619          STA  NEWARYTAB
ØE57:           620 ;
ØE57:           621 ;     THE MOVE ROUTINE:  TWO ARRAYS HAVE SAME NAME.  THE
ØE57:           622 ;     WRONG ONE IS LOWER IN MEMORY.  MOVE RENAMED ARRAY
ØE57:           623 ;     DOWN TO THE START OF ARRAY STORAGE, VIA THE INPUT
ØE57:           624 ;     BUFFER AT $200, IN SEGMENTS NO LONGER THAN 1 PAGE.
ØE57:           625 ;     MAKE ROOM FOR IT BY MOVING OLD ARRAYS UP UNTIL THEY
ØE57:           626 ;     OVERWRITE THE SEGMENT OF THE ARRAY MOVED DOWN.  NO
ØE57:           627 ;     VARIABLE STORAGE SPACE IS USED BY THIS ROUTINE, SO
ØE57:           628 ;     IF GRAPHICS ARE STORED HIGHER, THEY ARE LEFT ALONE.
ØE57:           629 ;
ØE57:AD 17 Ø8   630 MOVEARY  LDA  NEWARYTAB  ; POINTS TO FIRST ARRAY
ØE5A:85 6B      631          STA  ARYTAB     ; FOLLOWING THE LAST ONE
ØE5C:AD 18 Ø8   632          LDA  NEWARYTAB+1 ; THAT WAS MOVED DOWN
ØE5F:85 6C      633          STA  ARYTAB+1
ØE61:BD 1A Ø8   634          LDA  SECBUF,X   ; FIND NEXT ARRAY PTR
ØE64:85 FA      635          STA  GENPTR     ; SAVE IT
ØE66:E8         636          INX             ; GET THE HIGH BYTE
```

LISTING 1: SUBR.MASTER *(continued)*

```
0E67:BD 1A 08  637          LDA  SECBUF,X
0E6A:85 FB     638          STA  GENPTR+1
0E6C:E8        639          INX              ; POINT TO NEXT ARRAY
0E6D:86 DA     640          STX  BUFPTR      ; SAVE IT
0E6F:A0 02     641          LDY  #2          ; FETCH OFFSET OF
0E71:38        642          SEC              ; THIS ARRAY TO NEXT
0E72:B1 FA     643          LDA  (GENPTR),Y  ; THIS POINTS TO
0E74:AA        644          TAX              ; START OF NEXT ARRAY.
0E75:E9 01     645          SBC  #1          ; THIS POINTS TO END OF
0E77:85 77     646          STA  NUMCHR      ; THIS ONE.
0E79:C8        647          INY              ; FETCH HIGH BYTE
0E7A:B1 FA     648          LDA  (GENPTR),Y  ; # FULL PAGES TO MOVE
0E7C:48        649          PHA              ; SAVE TO ADJUST ARYTAB
0E7D:E9 00     650          SBC  #0          ; CARRY FROM -1 ABOVE
0E7F:85 78     651          STA  NUMPAGE     ; NOW ADJUST THIS UP
0E81:E6 78     652          INC  NUMPAGE     ; BY 1 FOR THE PARTIAL PAGE
0E83:18        653          CLC              ; ADD OFFSET OF ARRAY
0E84:8A        654          TXA              ; TO ARYTAB TO GET FIRST
0E85:65 6B     655          ADC  ARYTAB      ; LOC OF NEXT ARRAY
0E87:8D 17 08  656          STA  NEWARYTAB   ; ONCE THIS ONE
0E8A:68        657          PLA              ; IS MOVED TO THE BOTTOM
0E8B:65 6C     658          ADC  ARYTAB+1    ; OF ARRAY STORAGE
0E8D:8D 18 08  659          STA  NEWARYTAB+1
0E90:A4 77     660          LDY  NUMCHR      ; IS THERE A PARTIAL PAGE?
0E92:D0 04     661          BNE  MOVE1       ; YES, MOVE IT
0E94:C6 78     662          DEC  NUMPAGE     ; NO, THEN DONE PARTIAL PAGE
0E96:C6 77     663          DEC  NUMCHR      ; AND SET THIS TO #$FF
0E98:A5 6B     664 MOVE1    LDA  ARYTAB      ; SET UP 1ST BLTU
0E9A:85 9B     665          STA  LOWTR       ; MOVE FROM BOTTOM OF
0E9C:A5 6C     666          LDA  ARYTAB+1    ; OLD ARRAYS TO OVERWRITE
0E9E:85 9C     667          STA  LOWTR+1     ; ARRAY SEGMENT BEING MOVED
0EA0:B1 FA     668 MOVEDOWN LDA  (GENPTR),Y  ; TO THE BUFFER
0EA2:99 00 02  669          STA  BUFR,Y      ; MOVE THIS SEGMENT DOWN
0EA5:88        670          DEY              ; TO PAGE $200-$2FF, IE
0EA6:D0 F8     671          BNE  MOVEDOWN    ; THE INPUT BUFFER
0EA8:B1 FA     672          LDA  (GENPTR),Y  ; COVERS Y=0
0EAA:8D 00 02  673          STA  BUFR        ; NOW MOVE BUFFER UP.
0EAD:38        674          SEC              ; ADDS 1 TO # OF BYTES, AS
0EAE:A5 FA     675          LDA  GENPTR      ; REQUIRED FOR BLTU. MOVE
0EB0:85 96     676          STA  HIGHTR      ; FROM LOW TRANSFER ADDRESS
0EB2:65 77     677          ADC  NUMCHR      ; (LOWTR) UP, WITH LAST BYTE
0EB4:85 94     678          STA  HIGHDS      ; STORED IN HIGH DESTINATION
0EB6:A5 FB     679          LDA  GENPTR+1    ; WHICH IS HIGHDS -1
0EB8:85 97     680          STA  HIGHTR+1    ; MOVE ONLY A PAGE.  DON'T ADD 1
0EBA:69 00     681          ADC  #0          ; FOR HIGHTR, OR WILL MOVE
0EBC:85 95     682          STA  HIGHDS+1    ; TOO MUCH.
0EBE:20 9A D3  683          JSR  BLTUP       ; LATE ENTRY AVOIDS REASON CHECK
0EC1:A4 77     684          LDY  NUMCHR      ; NOW MOVE BUFR UP
0EC3:B9 00 02  685 MOVEUP   LDA  BUFR,Y      ; TO EMPTY AREA
0EC6:91 9B     686          STA  (LOWTR),Y   ; ABOVE OLD LOW
0EC8:88        687          DEY              ; TRANSFER ADDRESS
0EC9:D0 F8     688          BNE  MOVEUP      ; AGAIN, MISSES Y=0
0ECB:AD 00 02  689          LDA  BUFR        ; SO DO IT HERE
0ECE:91 9B     690          STA  (LOWTR),Y   ; SINCE 0 BYTE MOVED
0ED0:38        691          SEC              ; ACTUALLY MOVED NUMCHR+1 BYTES
0ED1:A5 9B     692          LDA  LOWTR       ; CARRY SET ADDS 1 MORE
0ED3:65 77     693          ADC  NUMCHR      ; THIS CALCULATES THE NEW ADDRESS
0ED5:85 9B     694          STA  LOWTR       ; OF THE BOTTOM OF THE
0ED7:90 02     695          BCC  MOVE2       ; OLD ARRAYS TO MOVE UP.
0ED9:E6 9C     696          INC  LOWTR+1     ; IF NEEDED.
0EDB:38        697 MOVE2    SEC              ; FOR NUMCHR+1
0EDC:A5 FA     698          LDA  GENPTR      ; UPDATE PTR TO REMAINDER
0EDE:65 77     699          ADC  NUMCHR      ; OF ARRAY TO GO DOWN
0EE0:85 FA     700          STA  GENPTR
0EE2:90 02     701          BCC  MOVE3
0EE4:E6 FB     702          INC  GENPTR+1
0EE6:A0 FF     703 MOVE3    LDY  #$FF        ; FULL PAGE MOVES FROM
0EE8:84 77     704          STY  NUMCHR      ; HERE, Y SET FOR MOVEDOWN.
0EEA:C6 78     705          DEC  NUMPAGE     ; MORE PAGES TO MOVE?
0EEC:D0 B2     706          BNE  MOVEDOWN    ; YES, DO.
0EEE:A6 DA     707          LDX  BUFPTR      ; NO, MORE VARS TO MOVE?
0EF0:E4 DB     708          CPX  COUNTER
0EF2:B0 03     709          BCS  MOVED       ; NO, DONE ARRAY PASS.
0EF4:4C 57 0E  710          JMP  MOVEARY     ; YES, DO NEXT.
0EF7:20 43 0A  711 MOVED    JSR  GETARYTAB   ; RECOVER ARYTAB
0EFA:20 10 0A  712 LOCAL    JSR  GETNAME     ; VARS ALL PASSED. PROC
0EFD:A0 04     713          LDY  #4          ; NAMED VAR HOLDS ADDRESS OF THIS
0EFF:B1 83     714 SAVEPROC LDA  (VARPNT),Y  ; PROGRAM. SAVE THE
0F01:99 1A 08  715          STA  SECBUF,Y    ; STORED REPRESENTATION, AND
0F04:88        716          DEY              ; LOAD IT INTO A NEW PROCNAME VAR
0F05:10 F8     717          BPL  SAVEPROC    ; LATER.  THIS ALLOWS RECURSION.
0F07:C8        718          INY              ; BACK TO 0
0F08:AD 15 08  719          LDA  OLDSIMPLE   ; STORE START ADDRESS OF
0F0B:91 83     720          STA  (VARPNT),Y  ; (Y=0) MAIN'S SIMPLES
0F0D:C8        721          INY              ; (Y=1) IN PROCNAME VAR
0F0E:AD 16 08  722          LDA  OLDSIMPLE+1
0F11:91 83     723          STA  (VARPNT),Y
0F13:20 05 0A  724          JSR  POINTNAME   ; MAKE A NEW VARIABLE
0F16:20 8E 0B  725          JSR  MAKESIMPLE  ; WITH PROC'S NAME.
0F19:A0 00     726          LDY  #0          ; PUT FURTHER RETURN DATA
0F1B:AD 07 08  727          LDA  DEFLIST     ; AWAY IN THE NEW
0F1E:91 83     728          STA  (VARPNT),Y  ; PROC VAR.
0F20:C8        729          INY              ; Y=1
0F21:AD 08 08  730          LDA  DEFLIST+1
0F24:91 83     731          STA  (VARPNT),Y  ; START OF DEFLIST SAVED
```

without properly exiting from the called routine. Local variables will now be global, and renamed variables will stay renamed.

Similarly, you might call a subroutine within a FOR loop and have one too many NEXTs inside it. You get back to the FOR without executing the EXIT. In the case of FOR, you can protect yourself easily. Use NEXT with the proper index variable (e.g., NEXT I). When the index is encountered, Applesoft checks it against the name it should find (saved on the stack) in the location it expects to find the index (also on the stack). The index variable has been moved by the creation of locals (at least three) during the CALL, so the test fails, and BASIC crashes the program with a NEXT WITHOUT FOR error — just as it does with crossed FORs and GOSUBs. However, if you don't specify the index variable, and you do cross the routines, you are in trouble. If you encounter this kind of problem, pressing <RESET> followed by FP reinitializes everything from DOS 3.3. Reload the handler and your program, fix the program and try again. From ProDOS you must reboot.

Crossing FOR and DISP — DISP is a subroutine of the handler that can be used in its own right to clear variables out of memory. If you clear a variable that occurred lower in memory than the index of a FOR loop, while the loop is active, and don't specify the index in the NEXT, the computer will create and clear the variables in the DISP list until you stop the program with <CTRL>C or <RESET>. If you specify the index (NEXT I instead of NEXT), the problem will be flagged with an error message. Further, the problem is rare because typical index variables, like I, are usually defined very early in the program, before variables that you would want to clear out later, so their location is not affected by the clear.

ONERR-GOTO, ONERR-GOODBYE — ONERR should be used with extreme caution, or not at all, when calling or exiting a subroutine. Within the CALL and EXIT, Applesoft's internal pointers are readjusted in unconventional ways when making and clearing local variables. If a programming error is detected at this stage (very rare since most errors are detected before new local variables are made), your pointers on return are not valid.

Modified Program Code — In a few places in our program, we have to look at the Applesoft program's variables one at a time. A comma after a variable is usually an adequate separator, but not for a DIM or a DISP, which expect variable lists. In these cases, we temporarily trade the comma for a colon. If an array in the LOCAL list is syntactically mis-specified (e.g., D(−2)), BASIC crashes while DIMming it before we

can put the comma back. Thus, the program now has a colon where we used to have a comma. This is useful for pinpointing the error: if there is a new colon, you know that the variable preceding it is the bad one. Further, it's harmless. If you miss it, you get a SYNTAX ERROR next time. Still, it modifies the code, which we didn't intend.

USEFUL HANDLER SUBROUTINES

Dispose

At last, we have come to dispose! Dispose is a Pascal command (Jensen and Wirth Standard, i.e., the original Pascal) that some microcomputer software distributors (like Apple) did not include in their versions of Pascal. If you set DISP=2304 (DISPOSE is parsed by BASIC into DIS POS E), and CALL DISP,*variable list*, all of the variables in the list will be erased from memory.

DISP can also do strange things to DEF FN functions, just as it can do them to FOR loops, as noted previously. If you DEF an FN after declaring a variable which you then dispose of, the FN will be moved down, and its internal pointers will be incorrect. Disposing of array variables never affects FNs, nor does clearing of simple variables that first appeared in the program after the DEF FN statement. If you use FNs in your main program, be cautious with DISP.

The remaining routines are only of interest to assembly language programmers. We assume that you have the premier issue of *Apple Orchard*, with Crossley's documentation of Applesoft pointers and subroutine, and that you have one of Apple's reference manuals.

NEWMOVE

The NEWMOVE routine starting at $96F mimics the Monitor MOVE routine. Its inputs are the same and it leaves Carry Set on exit, as does MOVE. It may repeat sequences if used to move data upward in memory, just as MOVE does. To move data up without worrying about this, use BLTU (Block Transfer Up) at $D393 or BLTUP at $D39A, which doesn't check or change STREND (so use this cautiously).

The Monitor MOVE ($FE2C) is documented on pages 44-46 and 55-56 of the *Apple II Reference Manual*. It moves data starting at the address pointed to by $3C,$3D through $3E,$3F, into the memory range starting at the location held in $42,$43. Our program does the same.

There are four differences between our routine and the Monitor's. If you specify a move starting location that is greater than or equal to the move ending location, our routine assumes you didn't mean it and gives an ILLEGAL QUANTITY error. The Monitor's MOVE moves nothing instead, or one byte, without flagging the error. Second, our program is longer than MOVE. Third, it's much faster for moves of more than a page (255 bytes) of data. For very small moves, MOVE is faster, but these take such a short

LISTING 1: SUBR.MASTER *(continued)*

```
0F26:C8        732          INY             ; Y=2
0F27:AD 09 08  733          LDA  CALLIST    ; SAVE START OF CALL
0F2A:91 83     734          STA  (VARPNT),Y ; VARIABLE LIST.
0F2C:C8        735          INY             ; Y=3
0F2D:AD 0A 08  736          LDA  CALLIST+1
0F30:91 83     737          STA  (VARPNT),Y
0F32:C8        738          INY             ; Y=4. LAST THIS VARIABLE.
0F33:A5 75     739          LDA  CURLIN     ; CALL STATEMENT
0F35:91 83     740          STA  (VARPNT),Y ; LINE NUMBER.
0F37:20 05 0A  741          JSR  POINTNAME  ; MAKE A NEW PROCNAME
0F3A:20 8E 0B  742          JSR  MAKESIMPLE ; TO HOLD THE REST.
0F3D:A0 04     743          LDY  #4
0F3F:A5 76     744          LDA  CURLIN+1   ; HIGH BYTE
0F41:91 83     745          STA  (VARPNT),Y
0F43:88        746          DEY             ; Y=3. GO DOWN TO 0
0F44:A5 6D     747          LDA  STREND     ; SNEAKY TRICK.
0F46:91 83     748          STA  (VARPNT),Y ; CHECK STREND WHEN
0F48:88        749          DEY             ; (Y=2) EXIT ATTEMPTED. IF THAT
0F49:A5 6E     750          LDA  STREND+1   ; STREND DOESN'T MATCH
0F4B:91 83     751          STA  (VARPNT),Y ; THIS ONE, GLOBAL VARS
0F4D:88        752          DEY             ; (Y=1) WERE TINKERED WITH. SCREAM.
0F4E:A9 00     753          LDA  #0         ; PUT 0 TO FLAG NO LOCAL LIST.
0F50:91 83     754          STA  (VARPNT),Y ; CHANGE IF IS ONE.
0F52:20 CC 09  755          JSR  POINTDEF   ; POINTS TO END OF LIST
0F55:20 72 0A  756          JSR  DEFTOCUR   ; PUT DEFLINE IN CURLIN
0F58:20 B7 00  757          JSR  CHRGOT     ; WHAT ENDS THE LIST?
0F5B:F0,03     758          BEQ  LOCAL0     ; MUST BE COLON OR EOL
0F5D:4C BC E1  759          JMP  DATAERR    ; OR LIST LONGER THAN CALL LIST.
0F60:C9 00     760 LOCAL0   CMP  #0         ; TRUE END OF LINE?
0F62:D0 16     761          BNE  LOCAL2     ; NO, WHAT FOLLOWS?
0F64:A0 02     762          LDY  #2         ; YES, FIND NEXT LINE
0F66:B1 B8     763          LDA  (TXTPTR),Y ; HIGH BYTE OF PTR
0F68:D0 03     764          BNE  LOCAL1     ; IF 0, END OF PROGRAM
0F6A:4C 71 0C  765          JMP  UNDEF      ; WHICH IS CRAZY.
0F6D:C8        766 LOCAL1   INY             ; POINT TO NEW LINE #
0F6E:B1 B8     767          LDA  (TXTPTR),Y ; SAVE IT AS
0F70:85 75     768          STA  CURLIN     ; CURRENT LINE
0F72:C8        769          INY
0F73:B1 B8     770          LDA  (TXTPTR),Y
0F75:85 76     771          STA  CURLIN+1
0F77:20 98 D9  772          JSR  ADDON      ; TXTPTR TO LINE'S TEXT
0F7A:A0 01     773 LOCAL2   LDY  #1         ; HAVE WE A "LOCAL" STATEMENT?
0F7C:B1 B8     774          LDA  (TXTPTR),Y ; LET'S SEE...
0F7E:C9 4C     775          CMP  #EL        ; L
0F80:D0 1C     776          BNE  NOTLOCAL
0F82:C8        777          INY
0F83:B1 B8     778          LDA  (TXTPTR),Y
0F85:C9 4F     779          CMP  #OH        ; O
0F87:D0 15     780          BNE  NOTLOCAL
0F89:C8        781          INY
0F8A:B1 B8     782          LDA  (TXTPTR),Y
0F8C:C9 43     783          CMP  #CE        ; C
0F8E:D0 0E     784          BNE  NOTLOCAL
0F90:C8        785          INY
0F91:B1 B8     786          LDA  (TXTPTR),Y
0F93:C9 41     787          CMP  #EH        ; A
0F95:D0 07     788          BNE  NOTLOCAL
0F97:C8        789          INY
0F98:B1 B8     790          LDA  (TXTPTR),Y
0F9A:C9 4C     791          CMP  #EL        ; L
0F9C:F0 06     792          BEQ  LOCAL3     ; L.O.C.A.L. YUP! DO IT.
0F9E:20 72 0A  793 NOTLOCAL JSR  DEFTOCUR   ; NO. RESTORE DEF LINE
0FA1:4C C4 0F  794          JMP  CALLDONE   ; AND LEAVE.
0FA4:C8        795 LOCAL3   INY             ; TXTPTR TO CHR AFTER "LOCAL"
0FA5:20 98 D9  796          JSR  ADDON      ; ONCE THIS DONE
0FA8:20 BE DE  797          JSR  CHKCOM     ; MUST BE COMMA THERE.
0FAB:A5 B8     798          LDA  TXTPTR     ; SAVE TRUE START OF LOCAL
0FAD:A0 00     799          LDY  #0         ; VARIABLE LIST.
0FAF:91 83     800          STA  (VARPNT),Y ; IN THE FIRST TWO BYTES
0FB1:C8        801          INY             ; OF THE LATEST PROCNAME VAR.
0FB2:A5 B9     802          LDA  TXTPTR+1   ; NEVER 0
0FB4:91 83     803          STA  (VARPNT),Y
0FB6:20 19 0A  804          JSR  DECTXT     ; MOVE BACK TO COMMA
0FB9:20 80 0B  805 LOCAL4   JSR  MAKEVAR    ; MAKE A LOCAL
0FBC:20 DE 09  806          JSR  TXTTODEF   ; UPDATE DEF LIST PTR
0FBF:20 B7 00  807          JSR  CHRGOT     ; END OF LIST?
0FC2:D0 F5     808          BNE  LOCAL4     ; NO, MAKE NEXT LOCAL
0FC4:20 05 0A  809 CALLDONE JSR  POINTNAME  ; ONE LAST PROCNAME
0FC7:20 8E 0B  810          JSR  MAKESIMPLE ; COMING UP.
0FCA:A0 04     811          LDY  #4         ; PUT THE CALL HANDLER
0FCC:B9 1A 08  812 LASTPROC LDA  SECBUF,Y   ; ADDRESS IN IT.
0FCF:91 83     813          STA  (VARPNT),Y
0FD1:88        814          DEY
0FD2:10 F8     815          BPL  LASTPROC
0FD4:20 CC 09  816          JSR  POINTDEF   ; POINT TO END OF DEF
0FD7:4C B8 09  817          JMP  OUT        ; RESTORE $FA-FF & EXIT
0FDA:          818 ********************************
0FDA:          819 *        CALL THIS ADDRESS     *
0FDA:          820 *        TO EXIT FROM PROC     *
0FDA:          821 ********************************
0FDA:20 AA 09  822 EXIT     JSR  IN         ; FREE UP $FA-FF
0FDD:A5 B8     823          LDA  TXTPTR     ; POINTS AT COMMA
0FDF:8D 0D 08  824          STA  PROCNAME   ; PRECEDING THE PROC
0FE2:A5 B9     825          LDA  TXTPTR+1   ; NAME FOLLOWING "EXIT"
0FE4:8D 0E 08  826          STA  PROCNAME+1
0FE7:20 BE DE  827          JSR  CHKCOM     ; BETTER BE A COMMA
```

time that it doesn't matter which routine you use. Finally, our routine sets Y internally. You need not set Y=0 before entering it.

Variable Finding Routines

Applesoft uses two subroutines to find the name and address of variables in memory: PTRGET ($DFE3) and GETARYPTR ($F7D9), which uses part of PTRGET. PTRGET finds the name of the variable (of any type), sets various flags in the process, and creates the variable if it didn't previously exist. It is this routine that DIMs previously undefined arrays. GETARYPTR finds the name of arrays only, sets flags, and looks for an array. It will not create a new array, but it will crash if one doesn't exist.

We needed to separate these functions. The subroutine GETVARNAM ($AE4) stores the name of your variable in LASTVAR ($81,$82) in the same way that PTRGET does, and it sets the same flags. It also flags simple versus array variables in ARYFLAG ($DD, an alias of ERRPOS), and expressions versus simple variables in EXPRFLAG ($DE, alias ERRNUM). Both error locations are used only when BASIC crashes, so you're not hurting anything by using them in the meantime. In the event of an Applesoft error, ARYFLAG and EXPRFLAG are overwritten. If you work with an array expression, ARYFLAG is set (holds $FF) if the first part of the expression is an array name. Otherwise, EXPRFLAG is set (holds $FF). GETVARNAM assumes that TXTPTR ($B8,$B9) points to the first character preceding the variable or expression. TXTPTR is unaffected by the routine.

> The DEF statement's variable list tells you automatically what types of variables the subroutine expects as input.

Subroutine FINDARY assumes that GETVARNAM has just been run. It searches memory for the array whose name is stored in LASTVAR. If the array exists, then, like GETARYPTR, it returns the first location of the array in LOWTR ($9B,$9C). Y is not zero in this case. If the array is not found, then rather than crashing OUT OF DATA (GETARYPTR) or DIMming an array (PTRGET), FINDARY returns with Y=0 and lets you decide what to do from here. TXTPTR is unaffected by the routine as is VARPNT ($83,$84), which PTRGET, but not GETARYPTR, changes.

Subroutine SKIPVAR ($A7D) will bypass an expression or variable in a list, with variables separated by commas or terminated by colons or the end of the line. On entry,

LISTING 1: SUBR.MASTER *(continued)*

```
0FEA:20 E3 DF   828          JSR  PTRGET     ; WHERE IS IT?
0FED:A0 00      829          LDY  #0         ; MUST SAVE THE
0FEF:B1 83      830          LDA  (VARPNT),Y ; FIRST TWO BYTES
0FF1:48         831          PHA             ; OF THE ADDRESS OF
0FF2:C8         832          INY             ; PROC AS THESE WERE
0FF3:B1 83      833          LDA  (VARPNT),Y ; OVERWRITTEN
0FF5:48         834          PHA             ; MAKING ROOM FOR OLDSIMPLE.
0FF6:20 05 0A   835          JSR  POINTNAME  ; NOW GET RID OF
0FF9:20 00 09   836          JSR  DISPOSE    ; THIS VARIABLE.
0FFC:20 10 0A   837          JSR  GETNAME    ; AND GET NEXT WITH THIS NAME
0FFF:A0 01      838          LDY  #1         ; DO WE HAVE A
1001:B1 83      839          LDA  (VARPNT),Y ; LOCAL VARLIST?
1003:F0 0D      840          BEQ  EXIT1      ; NOT IF THIS IS 0
1005:85 B9      841          STA  TXTPTR+1   ; IF YES, POINT TO IT
1007:88         842          DEY             ; POINTS TO TRUE START OF
1008:B1 83      843          LDA  (VARPNT),Y ; LOCAL LIST, NOT TO
100A:85 B8      844          STA  TXTPTR     ; LEADING COMMA. USE CLEAR.
100C:20 03 09   845          JSR  CLEAR      ; AND BYE, BYE LOCALS.
100F:20 10 0A   846          JSR  GETNAME    ; RECOVER PROCNAM LOC
1012:A0 02      847 EXIT1    LDY  #2         ; NOW CHECK STREND
1014:B1 83      848          LDA  (VARPNT),Y ; HIGH BYTE
1016:C5 6E      849          CMP  STREND+1   ; MUST MATCH
1018:D0 07      850          BNE  MEMORYERR
101A:C8         851          INY
101B:B1 83      852          LDA  (VARPNT),Y ; Y=3
101D:C5 6D      853          CMP  STREND
101F:F0 05      854          BEQ  EXIT3
1021:A2 54      855 MEMORYERR LDX #84        ; GLOBALS WERE TINKERED WITH.
1023:4C 12 D4   856          JMP  ERROR      ; "MEMORY ERROR"
1026:C8         857 EXIT3    INY             ; Y=4
1027:B1 83      858          LDA  (VARPNT),Y ; FETCH CALL LINE #
1029:85 76      859          STA  CURLIN+1   ; AND FLAG THIS AS CURRENT.
102B:20 05 0A   860          JSR  POINTNAME  ; THIS PROCNAME DONE
102E:20 00 09   861          JSR  DISPOSE    ; GOODBYE
1031:20 10 0A   862          JSR  GETNAME    ; NEXT PROCNAME, PLEASE.
1034:A0 04      863          LDY  #4         ; FETCH CURLIN LOW BYTE
1036:B1 83      864          LDA  (VARPNT),Y
1038:85 75      865          STA  CURLIN
103A:88         866          DEY             ; Y=3
103B:B1 83      867          LDA  (VARPNT),Y ; PTR TO
103D:8D 0A 08   868          STA  CALLIST+1  ; CALL VARLIST
1040:88         869          DEY             ; Y=2
1041:B1 83      870          LDA  (VARPNT),Y
1043:8D 09 08   871          STA  CALLIST
1046:88         872          DEY             ; Y=1
1047:B1 83      873          LDA  (VARPNT),Y ; PTR TO
1049:8D 08 08   874          STA  DEFLIST+1  ; DEF VARLIST
104C:88         875          DEY             ; Y=0
104D:B1 83      876          LDA  (VARPNT),Y
104F:8D 07 08   877          STA  DEFLIST
1052:20 05 0A   878          JSR  POINTNAME  ; ALL INFO FROM THIS
1055:20 00 09   879          JSR  DISPOSE    ; VAR USED. CLEAR IT.
1058:20 10 0A   880          JSR  GETNAME    ; FIND LAST ONE.
105B:A0 01      881          LDY  #1         ; AND RESTORE
105D:B1 83      882          LDA  (VARPNT),Y ; TOP TWO BYTES.
105F:8D 16 08   883          STA  OLDSIMPLE+1
1062:68         884          PLA             ; ADDRESS RECOVERY
1063:91 83      885          STA  (VARPNT),Y
1065:88         886          DEY             ; Y=0
1066:B1 83      887          LDA  (VARPNT),Y ; THEN START PASSING
1068:8D 15 08   888          STA  OLDSIMPLE  ; THE DATA BACK.
106B:68         889          PLA
106C:91 83      890          STA  (VARPNT),Y
106E:84 DA      891          STY  BUFPTR     ; INITIALIZE
1070:84 DB      892          STY  COUNTER    ; THE
1072:20 F0 09   893          JSR  STARTLIST  ; POINTERS
1075:20 CC 09   894 ARYBACK  JSR  POINTDEF   ; START BY PASSING
1078:20 B7 00   895          JSR  CHRGOT     ; ARRAYS, WHILE ARRAYS TO PASS.
107B:F0 33      896          BEQ  NAMEBACK   ; LIST THEM, THEN RENAME.
107D:20 E4 0A   897          JSR  GETVARNAM  ; GET NAME & TYPE
1080:A5 DD      898          LDA  ARYFLAG    ; GOT AN ARRAY?
1082:F0 27      899          BEQ  ABACK1     ; IF NOT, SKIP AND DO NEXT
1084:20 50 0B   900          JSR  FINDARY    ; GET ARRAY'S ADDRESS
1087:A6 DA      901          LDX  BUFPTR     ; SAVE NEW NAME, OLD ADDRESS.
1089:A5 9B      902          LDA  LOWTR      ; OLD ADDRESS
108B:9D 1C 08   903          STA  SECBUF+2,X ; THIS ORDER FOR
108E:A5 9C      904          LDA  LOWTR+1    ; THE RENAME ROUTINE
1090:9D 1D 08   905          STA  SECBUF+3,X
1093:20 D5 09   906          JSR  POINTCALL  ; GET THE NAME
1096:20 E4 0A   907          JSR  GETVARNAM  ; OF CALL LIST ARRAY
1099:A5 81      908          LDA  LASTVAR    ; NOW SAVE NEW NAME
109B:A6 DA      909          LDX  BUFPTR
109D:9D 1A 08   910          STA  SECBUF,X
10A0:A5 82      911          LDA  LASTVAR+1
10A2:9D 1B 08   912          STA  SECBUF+1,X
10A5:8A         913          TXA
10A6:18         914          CLC
10A7:69 04      915          ADC  #4
10A9:85 DA      916          STA  BUFPTR     ; UPDATE LIST POINTER
10AB:20 D2 0A   917 ABACK1   JSR  ADVANCEPTRS ; UPDATE VAR PTRS
10AE:90 C5      918          BCC  ARYBACK    ; ALWAYS TAKEN
10B0:20 22 0C   919 NAMEBACK JSR  RENAME
10B3:A9 00      920          LDA  #0
10B5:85 DA      921          STA  BUFPTR     ; INITIALIZE AGAIN
10B7:20 F0 09   922          JSR  STARTLIST  ; RESET CALL & DEF PTRS
```

TXTPTR points to the first character preceding the expression or variable. On exit, TXTPTR points to the first character (e.g., a comma) following it. This is the routine to use to skip variables or expressions after using GETVARNAM to find out what they are.

Local Variables

Subroutine MAKEVAR ($B80) creates all local variables. On entry, TXTPTR points to the character preceding the simple or array variable. If this is an expression that does not start with an array, you get a SYNTAX ERROR. If it is an array-started expression, whatever routine you call next will probably give you a SYNTAX ERROR, but don't bet on it. On exit, TXTPTR points to the character following the variable name. Simple variables are always created starting at the bottom of variable storage (VARTAB, $69,$6A). Arrays are created at the top of array storage (above what used to be storage end, $6D,$6E) if an array with that name does not exist. This is faster than creating one at the bottom of array storage (ARYTAB, $6B,$6C), which is only done if an old array has the same name as the new one.

Page Zero Save Routines

Many of these and other short utilities for moving page zero values are found at the start of the program ($9AA to $AAC). Specific locations from $801 up are used to store page zero variables while you use the page zero locations for some purpose of your own. The SECBUF region of page 8 (from $81A to $8FF) is used as temporary storage by various routines in this program but not by any you would be likely to use outside of this program. Thus, $81A to $8FF are free for temporary storage by your other assembly language subroutines. The subroutine handler does not rely on any values from this address range when the routine is entered, either at PROC (for CALLs) or at EXIT, so use these freely.

REFERENCES

Kaner, H.C. and J.R. Vokey. "Modifying Apple's Floating Point BASIC: An & Interpreter Without the &." *Compute!*, May 1982, pp. 146-152.

Mossberg, S. "LAMP — Part II." *Nibble*, Vol. 3/No. 3, 1982, pp. 33-39.

Mottola, R.M. "Amper-Interpreter." *Nibble*, Vol. 1/ No. 6, 1980, pp. 27-44.

Smith, M. "Using Named GOSUB and GOTO Statements in Applesoft BASIC." *Compute!*, May 1981, p. 64.

Worth, D., and P. Lechner. *Beneath Apple DOS*. Quality Software, 1981.

Yourdon, E. *Techniques of Program Structure and Design*. Prentice-Hall, 1975.

LISTING 1: SUBR.MASTER *(continued)*

```
10BA:20 CC 09  923 SIMPLEBACK JSR POINTDEF ; WHILE SIMPLES TO PASS
10BD:20 B7 00  924           JSR  CHRGOT   ; PASS THEM
10C0:F0 63     925           BEQ  LEAVE    ; THEN DONE
10C2:20 E4 0A  926           JSR  GETVARNAM ; WHAT HAVE WE?
10C5:A5 DD     927           LDA  ARYFLAG  ; IF 0, IS SIMPLE
10C7:D0 57     928           BNE  SBACK3   ; ELSE SKIP VARS
10C9:20 D5 09  929           JSR  POINTCALL ; FIND OUT WHAT CALL
10CC:20 E4 0A  930           JSR  GETVARNAM ; VAR IS BEFORE PASS BACK
10CF:A5 DE     931           LDA  EXPRFLAG ; DON'T WANT PASS TO EXPR
10D1:D0 33     932           BNE  SBACK2   ; IF EXPRESSION, PASS NOTHING
10D3:A5 DD     933           LDA  ARYFLAG  ; NOT 0 MEANS ARRAY OR ARRAY EXPR
10D5:F0 0F     934           BEQ  SBACK1   ; 0 IS SIMPLE, DO IT.
10D7:20 B1 00  935           JSR  CHRGET   ; ARRAY FLAG SET
10DA:20 E3 DF  936           JSR  PTRGET   ; MOVES TO END OF ARRAY
10DD:20 B7 00  937           JSR  CHRGOT   ; ARRAY IF FOLLOWED BY COMMA OR EOL
10E0:F0 04     938           BEQ  SBACK1   ; IN WHICH CASE PASS BACK
10E2:C9 2C     939           CMP  #COMMA   ; TO IT. ELSE ARRAY EXPR
10E4:D0 20     940           BNE  SBACK2   ; DON'T PASS IT
10E6:20 64 0A  941 SBACK1    JSR  SIMPTOVAR ; WILL PASS TO OLD SIMPLE
10E9:20 D5 09  942           JSR  POINTCALL ; CALL VAR OR TO CALL ARRAY
10EC:20 B1 00  943           JSR  CHRGET   ; PAST COMMA
10EF:20 E3 DF  944           JSR  PTRGET   ; FIND IT
10F2:A5 83     945           LDA  VARPNT   ; NOW PREPARE FOR LET
10F4:85 85     946           STA  FORPNT
10F6:A5 84     947           LDA  VARPNT+1
10F8:85 86     948           STA  FORPNT+1
10FA:20 2D 0A  949           JSR  GETVARTAB ; ALLOW FIND OF DEF SIMPLES
10FD:20 CC 09  950           JSR  POINTDEF ; SET CALL VAR = DEF VAR
1100:20 B1 00  951           JSR  CHRGET   ; SKIP COMMA
1103:20 52 DA  952           JSR  LETCNT   ; DO THE LET.
1106:20 CC 09  953 SBACK2    JSR  POINTDEF ; CLEAR OUT DEF SIMPLE.
1109:20 AC 0A  954           JSR  PUTCOLON ; ONLY CLEAR THE ONE VAR.
110C:20 00 09  955           JSR  DISPOSE
110F:20 CA 0A  956           JSR  REPCOLON ; RETURN THE CHR AFTER THE VAR
1112:38        957           SEC           ; AND UPDATE
1113:AD 15 08  958           LDA  OLDSIMPLE ; PTR TO THE
1116:E9 07     959           SBC  #7       ; OLD SIMPLE VARS
1118:8D 15 08  960           STA  OLDSIMPLE ; REFERRED TO
111B:B0 03     961           BCS  SBACK3   ; IN THE CALL LIST
111D:CE 16 08  962           DEC  OLDSIMPLE+1
1120:20 D2 0A  963 SBACK3    JSR  ADVANCEPTRS ; MOVE TO NEXT VAR
1123:90 95     964           BCC  SIMPLEBACK ; AND DO IT.
1125:20 D5 09  965 LEAVE     JSR  POINTCALL ; POINTS TO END OF CALL
1128:4C B8 09  966           JMP  OUT      ; RESTORE FA-FF & BACK TO BASIC
```

END OF LISTING 1

```
          KEY  PERFECT
            RUN ON
          SUBR.MASTER
==================================
CODE-5.0   ADDR# - ADDR#   CODE-4.0
--------   -------------   --------
448FE39B    0900 - 094F        2866
B643E5E9    0950 - 099F        261A
F24E6F74    09A0 - 09EF        2942
A3A23CE3    09F0 - 0A3F        2909
55D08203    0A40 - 0A8F        24DA
3211176C    0A90 - 0ADF        2707
F8DFD3E3    0AE0 - 0B2F        28B5
F69D94B0    0B30 - 0B7F        2A05
967B0F51    0B80 - 0BCF        264A
E8C00353    0BD0 - 0C1F        2802
47C956F4    0C20 - 0C6F        2B0C
1DBD82C5    0C70 - 0CBF        22AC
6A9ADDF7    0CC0 - 0D0F        286A
5AFE6098    0D10 - 0D5F        2B34
BB647EAD    0D60 - 0DAF        2410
A673FFB7    0DB0 - 0DFF        28B6
AD353C4E    0E00 - 0E4F        2873
624771CA    0E50 - 0E9F        2B1A
790312FB    0EA0 - 0EEF        2AB7
2326E17C    0EF0 - 0F3F        2609
8F729F86    0F40 - 0F8F        286C
28EC56AB    0F90 - 0FDF        25B1
31302CF4    0FE0 - 102F        25F6
5170D012    1030 - 107F        2680
132F20D9    1080 - 10CF        260D
1453E41A    10D0 - 111F        27E2
61ABAF59    1120 - 112A        0635
5BFD2671 = PROGRAM TOTAL =     082B
```

LISTING 2: SUBR.MAST.DEMO1

```
10  REM ************************
20  REM *  SUBR.MAST.DEMO1     *
30  REM * COPYRIGHT (C) 1985 *
40  REM * BY MICROSPARC, INC *
50  REM * CONCORD, MA  01742 *
60  REM ************************
70  IF  PEEK (104) <  > 17 THEN  POKE 103,44:
      POKE 104,17: POKE 4395,0: PRINT  CHR$ (
    4)"RUN SUBR.MAST.DEMO1"
80  IF  PEEK (2304) <  > 32 THEN  PRINT  CHR$
    (4)"BLOAD SUBR.MASTER"
90 SUB1 = 3141:RET = SUB1:EXIT = 4058
100  HOME : VTAB 12: HTAB 3: PRINT "DEMONSTRA
    TION OF SUBROUTINE MASTER": HTAB 6: PRINT
    "BY CEM KANER AND JOHN VOKEY": PRINT " C
    OPYRIGHT (C) 1985 BY MICROSPARC, INC.": CALL
    RET,"TO CONTINUE": HOME
110  HOME : INVERSE : PRINT "DEMONSTRATION OF
     PARAMETER PASSING": NORMAL
```

LISTING 2: SUBR.MAST.DEMO1 *(continued)*

```
120  PRINT : PRINT "THE VALUES OF THE VARIABL
     ES IN THE": PRINT "CALL STATEMENT ARE PA
     SSED TO THE": PRINT "CORRESPONDING VARIA
     BLES IN THE": PRINT "DEF STATEMENT:": LIST
     350: LIST 390
130  PRINT : PRINT "THE VALUES OF THE VARIABL
     ES IN THE": PRINT "DEF STATEMENT ARE PAS
     SED BACK TO THE": PRINT "CORRESPONDING V
     ARIABLES IN THE": PRINT "CALL STATEMENT.
     "
140  CALL RET,"FOR LISTING": HOME : INVERSE :
      PRINT "LISTING OF PARAMETER PASSING DEM
     O:": NORMAL : PRINT : LIST 330,430
150  CALL RET,"TO RUN PROGRAM": HOME : INVERSE
     : PRINT "PARAMETER PASSING DEMO": NORMAL
     : GOSUB 330: CALL RET,"FOR NEXT DEMO"
160  HOME : INVERSE : PRINT "DEMONSTRATION OF
      LOCAL VARIABLES": NORMAL
170  PRINT : PRINT "EACH VARIABLE IN THE DEF
     STATEMENT": PRINT "IS A LOCAL VARIABLE,
     DISTINCT FROM": PRINT "VARIABLES OF THE
     SAME NAME IN THE MAIN": PRINT "PROGRAM."
180  PRINT : PRINT "THE LOCAL STATEMENT CREAT
     ES ADDITIONAL": PRINT "LOCAL VARIABLES T
     HAT ARE DISTINCT": PRINT "FROM MAIN PROG
     RAM VARIABLES."
190  LIST 510
200  CALL RET,"TO LIST PROGRAM": HOME : INVERSE
     : PRINT "LISTING OF DEMO2": NORMAL : LIST
     440,540
210  CALL RET,"TO RUN PROGRAM": HOME : INVERSE
     : PRINT "LOCAL VARIABLE DEMO": NORMAL : GOSUB
     440
220  CALL RET,"FOR NEXT DEMO"
230  HOME : INVERSE : PRINT "EXPRESSION PASSI
     NG DEMO": NORMAL : PRINT : PRINT "EXPRES
     SIONS MAY BE USED IN THE": PRINT "CALL S
     TATEMENT:": LIST 570
240  PRINT "VARIABLES INCLUDED IN EXPRESSIONS
     ": PRINT "ARE NOT AFFECTED, EVEN IF THE"
     : PRINT "SUBROUTINE CHANGES THE VALUE OF
      THE": PRINT "CORRESPONDING VARIABLE IN
     THE": PRINT "DEF STATEMENT."
250  CALL RET,"TO LIST PROGRAM": HOME : INVERSE
     : PRINT "LISTING OF DEMO3": NORMAL : LIST
     550,650
260  CALL RET,"TO RUN PROGRAM": HOME : INVERSE
     : PRINT "EXPRESSION PASSING DEMO": NORMAL
     : GOSUB 550
270  CALL RET,"FOR NEXT DEMO"
280  HOME : INVERSE : PRINT "DEMONSTRATION OF
      PASSING STRINGS": NORMAL : PRINT : PRINT
     "STRING VARIABLES AND STRING LITERALS": PRINT
     "ARE HANDLED IN THE SAME WAY AS": PRINT
     "NUMERICS.": LIST 680: LIST 720
290  CALL RET,"FOR LISTING": HOME : INVERSE :
      PRINT "LISTING OF DEMO4": NORMAL : LIST
     660,760
300  CALL RET,"TO RUN PROGRAM": HOME : INVERSE
     : PRINT "STRING PASSING DEMO": NORMAL : GOSUB
     660
310  CALL RET,"TO QUIT": HOME
320  END
330  REM  PARAMETER PASSING DEMO
340 A = 5: PRINT : PRINT "A="A" BEFORE."
350  CALL SUB1,A
360  PRINT : PRINT "A="A" AFTER."
370  RETURN : REM *** RETURN FROM THIS DEMO
380  REM *** BEGINNING OF SUB1
390  DEF SUB1,N
400  PRINT : PRINT "N="N" (VALUE RECEIVED FRO
     M A)"
410 N = N * 10: PRINT : PRINT "VALUE OF N CHA
     NGED TO "N"."
420  CALL EXIT,SUB1
430  REM  *** END OF SUB1
440  REM  LOCAL VARIABLE DEMO
450 A = 14:B = 34: PRINT : PRINT "BEFORE:": PRINT
     "A="A" AND B="B"  (GLOBAL VARIABLES)
460  CALL SUB2,A
470  PRINT : PRINT "AFTER:": PRINT "A="A" AND
      B="B"  (GLOBAL VARIABLES)
480  RETURN : REM *** RETURN FROM THIS DEMO
490  REM *** BEGINNING OF SUB2
500  DEF SUB2,B
510  LOCAL,A
520 A = 4.52: PRINT : PRINT "DURING:": PRINT
     "A="A" AND B="B"  (LOCAL VARIABLES)
530  CALL EXIT,SUB2
540  REM *** END OF SUB2
550  REM  EXPRESSION DEMO
560 A = 2: PRINT : PRINT "BEFORE:": PRINT "A=
     "A
570  CALL SUB3,A + 5
580  PRINT : PRINT "AFTER:": PRINT "A="A
590  RETURN : REM  RETURN FROM THIS DEMO
600  REM *** BEGINNING OF SUB3
610  DEF SUB3,N
620  PRINT : PRINT "N="N"  (RECEIVED FROM MAI
     N)
630 N = N * 10: PRINT "N="N"  (CHANGED IN SUB
     3)
640  CALL EXIT,SUB3
650  REM *** END OF SUB3
660  REM  STRING DEMO
670 A$ = "ABC": PRINT : PRINT "BEFORE:": PRINT
     "A$=" CHR$ (34)A$ CHR$ (34)
680  CALL SUB4,A$
690  PRINT : PRINT "AFTER:": PRINT "A$=" CHR$
     (34)A$ CHR$ (34)
700  RETURN
710  REM *** BEGINNING OF SUB4
720  DEF SUB4,X$
730  PRINT : PRINT "X$=" CHR$ (34)X$ CHR$ (34
     )"  (RECEIVED BY SUB4) "
740 X$ = "-*-" + X$ + "-*-"
750  CALL EXIT,SUB4
760  REM *** END OF SUB4
770  DEF RET,MS$
780  LOCAL,Z$
790  VTAB 23: HTAB 1: PRINT "PRESS <RETURN> "
     MS$;: GET Z$: CALL EXIT,RET
```

END OF LISTING 2

```
                KEY  PERFECT
                  RUN ON
              SUBR.MAST.DEMO1
=======================================
   CODE-5.Ø   LINE# - LINE#   CODE-4.Ø
   --------   ------------   --------
   1AF77823      1Ø -   1ØØ      C8D1
   5C9EØCBE     11Ø -   2ØØ     Ø1759E
   917D3558     21Ø -   3ØØ     Ø14BF5
   7155AC8C     31Ø -   4ØØ       5A15
   BØ8A5Ø34     41Ø -   5ØØ       76CB
   EB1918CF     51Ø -   6ØØ       6458
   F2F1ADFB     61Ø -   7ØØ       64CA
   FDAØF4CB     71Ø -   79Ø       57A7
   476908AØ = PROGRAM TOTAL =     ØBA9
```

LISTING 3: SUBR.MAST.DEMO2

```
1Ø  REM ************************
2Ø  REM *  SUBR.MAST.DEMO2    *
3Ø  REM * COPYRIGHT (C) 1985 *
4Ø  REM * BY MICROSPARC, INC *
5Ø  REM * CONCORD, MA  Ø1742 *
6Ø  REM ************************
7Ø  IF  PEEK (1Ø4) <  > 17 THEN  POKE 1Ø3,44:
     POKE 1Ø4,17: POKE 4395,Ø: PRINT  CHR$ (
    4)"RUN SUBR.MAST.DEMO2"
8Ø  IF  PEEK (2Ø4) <  > 32 THEN  PRINT  CHR$
    (4)"BLOAD SUBR.MASTER"
9Ø FACT = 3141:EXIT = 4Ø58: HOME : VTAB 12: PRINT
     "FACTORIAL CALCULATIONS USING RECURSION"
     : PRINT : PRINT "* COPYRIGHT (C) 1985 BY
     MICROSPARC, INC*": VTAB 21: PRINT "PRES
     S <RETURN> TO START";: GET Z$: PRINT : HOME

1ØØ  INPUT "INPUT INTEGER (Ø TO 33): ";A
11Ø RS = 1: REM INITIALIZE RESULT TO 1
12Ø  CALL FACT,A
13Ø  PRINT RS
14Ø  GOTO 1ØØ
15Ø  REM *** BEGINNING OF FACT ROUTINE
16Ø  DEF FACT,N
17Ø  IF N > 1 THEN RS = RS * N: CALL FACT,N -
     1
18Ø  CALL EXIT,FACT
19Ø  REM *** END OF FACT ROUTINE
```

END OF LISTING 3